SAS/STAT® 14.3
User’s Guide
The GLMPOWER
Procedure
### Chapter 50
The GLMPOWER Procedure

#### Contents

<table>
<thead>
<tr>
<th>Section</th>
<th>Page</th>
</tr>
</thead>
<tbody>
<tr>
<td>Overview: GLMPOWER Procedure</td>
<td>3918</td>
</tr>
<tr>
<td>Getting Started: GLMPOWER Procedure</td>
<td>3919</td>
</tr>
<tr>
<td>Simple Two-Way ANOVA</td>
<td>3919</td>
</tr>
<tr>
<td>Incorporating Contrasts, Unbalanced Designs, and Multiple Means Scenarios</td>
<td>3923</td>
</tr>
<tr>
<td>Syntax: GLMPOWER Procedure</td>
<td>3925</td>
</tr>
<tr>
<td>PROC GLMPOWER Statement</td>
<td>3926</td>
</tr>
<tr>
<td>BY Statement</td>
<td>3927</td>
</tr>
<tr>
<td>CLASS Statement</td>
<td>3928</td>
</tr>
<tr>
<td>CONTRAST Statement</td>
<td>3928</td>
</tr>
<tr>
<td>MANOVA Statement</td>
<td>3929</td>
</tr>
<tr>
<td>MODEL Statement</td>
<td>3931</td>
</tr>
<tr>
<td>PLOT Statement</td>
<td>3931</td>
</tr>
<tr>
<td>POWER Statement</td>
<td>3936</td>
</tr>
<tr>
<td>REPEATED Statement</td>
<td>3943</td>
</tr>
<tr>
<td>WEIGHT Statement</td>
<td>3946</td>
</tr>
<tr>
<td>Details: GLMPOWER Procedure</td>
<td>3946</td>
</tr>
<tr>
<td>Specifying Value Lists in the POWER Statement</td>
<td>3946</td>
</tr>
<tr>
<td>Number-Lists</td>
<td>3947</td>
</tr>
<tr>
<td>Name-Lists</td>
<td>3947</td>
</tr>
<tr>
<td>Keyword-Lists</td>
<td>3947</td>
</tr>
<tr>
<td>Sample Size Adjustment Options</td>
<td>3947</td>
</tr>
<tr>
<td>Error and Information Output</td>
<td>3948</td>
</tr>
<tr>
<td>Displayed Output</td>
<td>3949</td>
</tr>
<tr>
<td>ODS Table Names</td>
<td>3950</td>
</tr>
<tr>
<td>Computational Methods and Formulas</td>
<td>3950</td>
</tr>
<tr>
<td>Contrasts in Fixed-Effect Univariate Models</td>
<td>3950</td>
</tr>
<tr>
<td>Adjustments for Covariates in Univariate Models</td>
<td>3952</td>
</tr>
<tr>
<td>Contrasts in Fixed-Effect Multivariate Models</td>
<td>3953</td>
</tr>
<tr>
<td>ODS Graphics</td>
<td>3961</td>
</tr>
<tr>
<td>Examples: GLMPOWER Procedure</td>
<td>3961</td>
</tr>
<tr>
<td>Example 50.1: One-Way ANOVA</td>
<td>3961</td>
</tr>
<tr>
<td>Example 50.2: Two-Way ANOVA with Covariate</td>
<td>3967</td>
</tr>
<tr>
<td>Example 50.3: Repeated Measures ANOVA</td>
<td>3974</td>
</tr>
<tr>
<td>References</td>
<td>3977</td>
</tr>
</tbody>
</table>
Overview: GLMPOWER Procedure

Power and sample size analysis optimizes the resource usage and design of a study, improving chances of conclusive results with maximum efficiency. The GLMPOWER procedure performs prospective power and sample size analysis for linear models, with a variety of goals:

- determining the sample size required to get a significant result with adequate probability (power)
- characterizing the power of a study to detect a meaningful effect
- conducting what-if analyses to assess sensitivity of the power or required sample size to other factors

Here *prospective* indicates that the analysis pertains to planning for a future study. This is in contrast to *retrospective* analysis for a past study, which is not supported by this procedure.

The statistical analyses that are covered include Type III $F$ tests and contrasts of fixed effects in univariate and multivariate linear models. For univariate models, you can specify covariates, which can be continuous or categorical. For multivariate models, you can choose among Wilks’ likelihood ratio, Hotelling-Lawley trace, and Pillai’s trace $F$ tests for multivariate analysis of variance (MANOVA) and among uncorrected, Greenhouse-Geisser, Huynh-Feldt, and Box conservative $F$ tests for the univariate approach to repeated measures. Tests and contrasts that involve random effects are not supported. For power and sample size analyses in a variety of other statistical situations, see Chapter 91, “The POWER Procedure.”

Input for PROC GLMPOWER includes the following components, which are considered in study planning:

- design (including subject profiles and their allocation weights)
- statistical model and test
- between-subject contrasts of class effects
- within-subject contrasts (for multivariate models)
- significance level (alpha)
- surmised response means for subject profiles (often called “cell means”)
- surmised variability (and correlation for multivariate models)
- power
- sample size

In order to identify power or sample size as the result parameter, you designate it by a missing value in the input. The procedure calculates this result value over one or more scenarios of input values for all other components.

You specify the design and the cell means by using an *exemplary data set*, a data set of artificial values that is constructed to represent the intended sampling design and the surmised response means in the underlying population. You specify the model and between-subject contrasts by using MODEL and CONTRAST statements similar to those in the GLM, ANOVA, and MIXED procedures. For multivariate models, you
specify the within-subject contrasts by using MANOVA and REPEATED statements similar to those in the GLM and MIXED procedures. You specify the remaining parameters by using the POWER statement, which is similar to analysis statements in the POWER procedure.

In addition to tabular results, PROC GLMPOWER produces graphs. You can produce the most common types of plots easily with default settings and use a variety of options for more customized graphics. For example, you can control the choice of axis variables, axis ranges, number of plotted points, mapping of graphical features (such as color, line style, symbol, and panel) to analysis parameters, and legend appearance.

If ODS Graphics is enabled, then PROC GLMPOWER uses ODS Graphics to create graphs; otherwise, traditional graphs are produced.

For more information about enabling and disabling ODS Graphics, see the section “Enabling and Disabling ODS Graphics” on page 615 in Chapter 21, “Statistical Graphics Using ODS.”

For specific information about the statistical graphics and options available with the GLMPOWER procedure, see the PLOT statement and the section “ODS Graphics” on page 3961.

The GLMPOWER procedure is one of several tools available in SAS/STAT software for power and sample size analysis. PROC POWER covers a variety of other analyses such as t tests, equivalence tests, confidence intervals, binomial proportions, multiple regression, one-way ANOVA, survival analysis, logistic regression, and the Wilcoxon rank-sum test. SAS Studio provides a user interface and implements many of the analyses supported in the procedures. For more information, see Chapter 91, “The POWER Procedure,” and SAS Studio: Task Reference Guide.

The following sections of this chapter describe how to use PROC GLMPOWER and discuss the underlying statistical methodology. The section “Getting Started: GLMPOWER Procedure” on page 3919 introduces PROC GLMPOWER with examples of power computation for a two-way analysis of variance. The section “Syntax: GLMPOWER Procedure” on page 3925 describes the syntax of the procedure. The section “Details: GLMPOWER Procedure” on page 3946 summarizes the methods employed by PROC GLMPOWER and provides details on several special topics. The section “Examples: GLMPOWER Procedure” on page 3961 illustrates the use of the GLMPOWER procedure with several applications.

For an overview of methodology and SAS tools for power and sample size analysis, see Chapter 18, “Introduction to Power and Sample Size Analysis.” For more discussion and examples for linear models, see Castelloe and O’Brien (2001); O’Brien and Shieh (1992); Muller and Benignus (1992); O’Brien and Muller (1993). For additional discussion of general power and sample size concepts, see O’Brien and Castelloe (2007); Castelloe (2000); Muller and Benignus (1992); Lenth (2001).

---

**Getting Started: GLMPOWER Procedure**

**Simple Two-Way ANOVA**

This example demonstrates how to use PROC GLMPOWER to compute and plot power for each effect test in a two-way analysis of variance (ANOVA).

Suppose you are planning an experiment to study the effect of light exposure at three levels on the growth of two varieties of flowers. The planned data analysis is a two-way ANOVA with flower height (measured
at two weeks) as the response and a model consisting of the effects of light exposure, flower variety, and their interaction. You want to calculate the power of each effect test for a balanced design with a total of 60 specimens (10 for each combination of exposure and variety) with $\alpha = 0.05$ for each test.

As a first step, create an exemplary data set describing your conjectures about the underlying population means. You believe that the mean flower height for each combination of variety and exposure level (that is, for each design profile, or for each cell in the design) roughly follows Table 50.1.

**Table 50.1** Mean Flower Height (in cm) by Variety and Exposure

<table>
<thead>
<tr>
<th>Variety</th>
<th>Exposure</th>
<th>1</th>
<th>2</th>
<th>3</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td></td>
<td>14</td>
<td>16</td>
<td>21</td>
</tr>
<tr>
<td>2</td>
<td></td>
<td>10</td>
<td>15</td>
<td>16</td>
</tr>
</tbody>
</table>

The following statements create a data set named Exemplary containing these cell means.

```sas
data Exemplary;
  do Variety = 1 to 2;
    do Exposure = 1 to 3;
      input Height @@;
      output;
    end;
  end;
datalines;
  14 16 21
  10 15 16
;```

You also conjecture that the error standard deviation is about 5 cm.

Use the **DATA=** option in the **PROC GLMPOWER** statement to specify Exemplary as the exemplary data set. Identify the classification variables (Variety and Exposure) by using the **CLASS** statement. Specify the model by using the **MODEL** statement. Use the **POWER** statement to specify power as the result parameter and provide values for the other analysis parameters, error standard deviation and total sample size. The following SAS statements perform the power analysis:

```sas
proc glmpower data=Exemplary;
  class Variety Exposure;
  model Height = Variety | Exposure;
  power
    stddev = 5
    ntotal = 60
    power = .;
run;
```

The **MODEL** statement defines the full model including both main effects and the interaction. The **POWER=** option in the **POWER** statement identifies power as the result parameter with a missing value (**POWER=**). The **STDDEV=** option specifies an error standard deviation of 5, and the **NTOTAL=** option specifies a total sample size of 60. The default value for the **ALPHA=** option sets the significance level to $\alpha = 0.05$.

Figure 50.1 shows the output.
Figure 50.1 Sample Size Analysis for Two-Way ANOVA

The GLMPOWER Procedure

<table>
<thead>
<tr>
<th>Fixed Scenario Elements</th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td>Dependent Variable</td>
<td>Height</td>
</tr>
<tr>
<td>Error Standard Deviation</td>
<td>5</td>
</tr>
<tr>
<td>Total Sample Size</td>
<td>60</td>
</tr>
<tr>
<td>Alpha</td>
<td>0.05</td>
</tr>
<tr>
<td>Error Degrees of Freedom</td>
<td>54</td>
</tr>
</tbody>
</table>

Computed Power

<table>
<thead>
<tr>
<th>Index</th>
<th>Source</th>
<th>Test DF</th>
<th>Power</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>Variety</td>
<td>1</td>
<td>0.718</td>
</tr>
<tr>
<td>2</td>
<td>Exposure</td>
<td>2</td>
<td>0.957</td>
</tr>
<tr>
<td>3</td>
<td>Variety*Exposure</td>
<td>2</td>
<td>0.191</td>
</tr>
</tbody>
</table>

The power is about 0.72 for the test of the Variety effect. In other words, there is a probability of 0.72 that the test of the Variety effect will produce a significant result (given the assumptions for the means and error standard deviation). The power is 0.96 for the test of the Exposure effect and 0.19 for the interaction test.

Now, suppose you want to account for some of your uncertainty in conjecturing the true error standard deviation by evaluating the power at reasonable low and high values, 4 and 6.5. You also want to plot power for sample sizes between 30 and 90. The following statements perform the analysis:

```r
ods graphics on;

proc glmpower data=Exemplary;
   class Variety Exposure;
   model Height = Variety | Exposure;
   power
       stddev = 4 6.5
       ntotal = 60
       power = .;
   plot x=n min=30 max=90;
run;

ods graphics off;
```

The PLOT statement with the X=N option requests a plot with sample size on the X axis. (The result parameter—in this case, power—is always plotted on the other axis.) The MIN= and MAX= options in the PLOT statement specify the sample size range. The ODS GRAPHICS ON statement enables ODS Graphics. Figure 50.2 shows the output, and Figure 50.3 shows the plot.
Figure 50.2 reveals that the power ranges from about 0.130 to 0.996 for the different effect tests and scenarios for standard deviation, with a sample size of 60. In Figure 50.3, the line style identifies the effect test, and the plotting symbol identifies the standard deviation. The locations of the plotting symbols identify actual computed powers; the curves are linear interpolations of these points. Note that the computed points in the plot occur at sample size multiples of 6, because there are 6 cells in the design (and by default, sample sizes are rounded to produce integer cell sizes).
Incorporating Contrasts, Unbalanced Designs, and Multiple Means Scenarios

Figure 50.3 Plot of Power versus Sample Size for Two-Way ANOVA with Input Ranges

Suppose you want to compute power for the two-way ANOVA described in the section “Simple Two-Way ANOVA” on page 3919, but you want to additionally perform the following tasks:

- try an unbalanced sample size allocation with respect to Exposure, using twice as many samples for levels 2 and 3 as for level 1
- consider an additional, less optimistic scenario for the cell means, shown in Table 50.2
- test a contrast of Exposure comparing levels 1 and 3
To specify the unbalanced design and the additional cell means scenario, you can add two new variables to the exemplary data set (Weight for the sample size weights, and HeightNew for the new cell means scenario). Change the name of the original cell means scenario to HeightOrig. The following statements define the exemplary data set:

```sas
data Exemplary;
  input Variety $ Exposure $ HeightOrig HeightNew Weight;
  datalines;
    1 1 14 15 1
    1 2 16 16 2
    1 3 21 20 2
    2 1 10 11 1
    2 2 15 14 2
    2 3 16 15 2
;
```

In PROC GLMPOWER, specify the name of the weight variable by using the WEIGHT statement, and specify the name of the cell means variables as dependent variables in the MODEL statement. Use the CONTRAST statement to specify the contrast as you would in PROC GLM. The following statements perform the sample size analysis.

```sas
proc glmpower data=Exemplary;
  class Variety Exposure;
  model HeightOrig HeightNew = Variety | Exposure;
  weight Weight;
  contrast 'Exposure=1 vs Exposure=3' Exposure 1 0 -1;
  power
    stddev = 5
    ntotal = 60
    power = .;
run;
```

Figure 50.4 shows the output.

**Figure 50.4** Sample Size Analysis for More Complex Two-Way ANOVA

### Table 50.2 Additional Cell Means Scenario

<table>
<thead>
<tr>
<th>Variety</th>
<th>Exposure</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>15 16 20</td>
</tr>
<tr>
<td>2</td>
<td>11 14 15</td>
</tr>
</tbody>
</table>

**The GLMPOWER Procedure**

<table>
<thead>
<tr>
<th>Fixed Scenario Elements</th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td>Weight Variable</td>
<td>Weight</td>
</tr>
<tr>
<td>Error Standard Deviation</td>
<td>5</td>
</tr>
<tr>
<td>Total Sample Size</td>
<td>60</td>
</tr>
<tr>
<td>Alpha</td>
<td>0.05</td>
</tr>
<tr>
<td>Error Degrees of Freedom</td>
<td>54</td>
</tr>
</tbody>
</table>
The power of the contrast of Exposure levels 1 and 3 is about 0.95 for the original cell means scenario (HeightOrig) and only 0.71 for the new one (HeightNew). The power is higher for the test of Variety, but lower for the tests of Exposure and of Variety*Exposure for the new cell means scenario compared to the original one. Note also for the HeightOrig scenario that the power for the unbalanced design (Figure 50.4) compared to the balanced design (Figure 50.1) is slightly lower for the tests of Variety and Exposure, but slightly higher for the test of Variety*Exposure.

**Syntax: GLMPOWER Procedure**

The following statements are available in the GLMPOWER procedure:

```plaintext
PROC GLMPOWER < options > ;
   BY variables ;
   CLASS variables ;
   CONTRAST 'label' effect values < . . . effect values > < / options > ;
   MANOVA 'label' < test-options > < / detail-options > ;
   MODEL dependent-variables = independent-effects ;
   PLOT < plot-options > < / graph-options > ;
   POWER < options > ;
   REPEATED factor-specification ;
   WEIGHT variable ;
```

The PROC GLMPOWER statement, the MODEL statement, and the POWER statement are required. If your model contains classification effects, the classification variables must be listed in a CLASS statement, and the CLASS statement must appear before the MODEL statement. In addition, CONTRAST and POWER statements must appear after the MODEL statement. PLOT statements must appear after the POWER statement that defines the analysis for the plot.

If you specify one or more MANOVA or REPEATED statements, then the model is assumed to be multivariate. Otherwise, a univariate model is assumed, in which case multiple dependent variables represent cell means scenarios for a single response.

You can use multiple CONTRAST, MANOVA, REPEATED, POWER, and PLOT statements. Each CONTRAST statement defines a separate between-subject contrast. Each MANOVA or REPEATED statement...
defines a separate within-subject contrast for a multivariate model. Each POWER statement produces a separate analysis and uses the information that is contained in the CLASS, MODEL, WEIGHT, CONTRAST, MANOVA, and REPEATED statements. Each PLOT statement refers to the previous POWER statement and generates a separate graph (or set of graphs).

Table 50.3 summarizes the basic functions of each statement in PROC GLMPOWER. The syntax of each statement in Table 50.3 is described in the following pages.

**Table 50.3  Statements in the GLMPOWER Procedure**

<table>
<thead>
<tr>
<th>Statement</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>PROC GLMPOWER</td>
<td>Invokes procedure and specifies exemplary data set</td>
</tr>
<tr>
<td>BY</td>
<td>Specifies variables to define subgroups for the analysis</td>
</tr>
<tr>
<td>CLASS</td>
<td>Declares classification variables</td>
</tr>
<tr>
<td>CONTRAST</td>
<td>Defines between-subject linear tests of model parameters</td>
</tr>
<tr>
<td>MANOVA</td>
<td>Defines within-subject linear tests of model parameters for multivariate models, in terms of contrast matrix coefficients</td>
</tr>
<tr>
<td>MODEL</td>
<td>Defines model and specifies dependent variables; for univariate models, multiple dependent variables represent cell means scenarios for a single response</td>
</tr>
<tr>
<td>PLOT</td>
<td>Displays graphs for preceding POWER statement</td>
</tr>
<tr>
<td>POWER</td>
<td>Identifies parameter to solve for and provides one or more scenarios for values of other analysis parameters</td>
</tr>
<tr>
<td>REPEATED</td>
<td>Defines within-subject linear tests of model parameters for multivariate models, in terms of common repeated measures transformations of the dependent variables</td>
</tr>
<tr>
<td>WEIGHT</td>
<td>Specifies variable for allocating sample sizes to different subject profiles</td>
</tr>
</tbody>
</table>

**PROC GLMPOWER Statement**

PROC GLMPOWER <options> ;

The PROC GLMPOWER statement invokes the GLMPOWER procedure. You can specify the following options.
**DATA=SAS-data-set**
names a SAS data set to be used as the exemplary data set, which is an artificial data set constructed to represent the intended sampling design and the conjectured response means for the underlying population.

**ORDER=DATA | FORMATTED | FREQ | INTERNAL**
specifies the sort order for the levels of the classification variables (which are specified in the CLASS statement).

This option applies to the levels for all classification variables, except when you use the (default) ORDER=FORMATTED option with numeric classification variables that have no explicit format. In that case, the levels of such variables are ordered by their internal value.

The ORDER= option can take the following values:

<table>
<thead>
<tr>
<th>Value of ORDER=</th>
<th>Levels Sorted By</th>
</tr>
</thead>
<tbody>
<tr>
<td>DATA</td>
<td>Order of appearance in the input data set</td>
</tr>
<tr>
<td>FORMATTED</td>
<td>External formatted value, except for numeric variables with no explicit format, which are sorted by their unformatted (internal) value</td>
</tr>
<tr>
<td>FREQ</td>
<td>Descending frequency count; levels with the most observations come first in the order</td>
</tr>
<tr>
<td>INTERNAL</td>
<td>Unformatted value</td>
</tr>
</tbody>
</table>

By default, ORDER=FORMATTED. For ORDER=FORMATTED and ORDER=INTERNAL, the sort order is machine-dependent.

For more information about sort order, see the chapter on the SORT procedure in the *SAS Visual Data Management and Utility Procedures Guide* and the discussion of BY-group processing in *SAS Language Reference: Concepts*.

**PLOTONLY**
specifies that only graphical results from the PLOT statement be produced.

---

**BY Statement**

```sas
BY variables ;
```

You can specify a BY statement with PROC GLMPOWER to obtain separate analyses of observations in groups that are defined by the BY variables. When a BY statement appears, the procedure expects the input data set to be sorted in order of the BY variables. If you specify more than one BY statement, only the last one specified is used.

If your input data set is not sorted in ascending order, use one of the following alternatives:

- Sort the data by using the SORT procedure with a similar BY statement.
- Specify the NOTSORTED or DESCENDING option in the BY statement for the GLMPOWER procedure. The NOTSORTED option does not mean that the data are unsorted but rather that the data are arranged in groups (according to values of the BY variables) and that these groups are not necessarily in alphabetical or increasing numeric order.

- Create an index on the BY variables by using the DATASETS procedure (in Base SAS software).

Because sorting the data changes the order in which PROC GLMPOWER reads observations, the sort order for the levels of the classification variables might be affected if you have also specified ORDER=DATA in the PROC GLMPOWER statement. This, in turn, affects specifications in CONTRAST statements.

For more information about BY-group processing, see the discussion in SAS Language Reference: Concepts. For more information about the DATASETS procedure, see the discussion in the SAS Visual Data Management and Utility Procedures Guide.

---

### CLASS Statement

```
CLASS variables;
```

The CLASS statement names the classification variables to be used in the analysis. If you use the CLASS statement, it must appear before the MODEL statement.

Classification variables can be either character or numeric. By default, class levels are determined from the entire set of formatted values of the CLASS variables.

### CONTRAST Statement

```
CONTRAST 'label' effect values <... effect values> </options> ;
```

The CONTRAST statement enables you to define custom Type III hypothesis tests by specifying an \( L \) vector or matrix for testing either the hypothesis \( L \beta = 0 \) (for univariate models) or the hypothesis \( LBM = 0 \) (for multivariate models). The \( L \) matrix consists of one or more between-subject contrasts.

To use this feature, you must be familiar with the details of the model parameterization that PROC GLM uses. For more information, see the section “Parameterization of PROC GLM Models” on page 3776 in Chapter 48, “The GLM Procedure.” All the elements of the \( L \) matrix can be given, or if only certain portions of the \( L \) matrix are given, PROC GLMPOWER constructs the remaining elements from the context (in a manner similar to that in rule 4 in the section “Construction of Least Squares Means” on page 3809 in Chapter 48, “The GLM Procedure”).

There is no limit to the number of CONTRAST statements that you can specify, but they must appear after the MODEL statement. Each power analysis includes tests for all CONTRAST statements.

You can specify the following arguments:

- **label** identifies the contrast in the output. A label is required for every contrast that is specified. Labels must be enclosed in single or double quotation marks.

- **effect** identifies an effect that appears in the MODEL statement, or the INTERCEPT effect. You do not need to include all effects that appear in the MODEL statement.
values are constants that are elements of the L matrix associated with the effect.

You can specify the following option in the CONTRAST statement after a slash (/):

SINGULAR=number

tunes the estimability checking. If \( \text{ABS}(L - LH) > C \times \text{number} \) for any row in the contrast, then L is declared nonestimable. H is the \((X'X)^{-1}X'\) matrix, and C is \(\text{ABS}(L)\) except for rows where L is zero, and then it is 1. The default value for the SINGULAR= option is \(10^{-4}\). Values for the SINGULAR= option must be between 0 and 1.

As stated previously, the CONTRAST statement enables you to define custom hypothesis tests. If the hypothesis is testable in a univariate model, then the hypothesis sum of squares, \(SS(H_0; L\beta = 0)\), is computed as

\[
(Lb)'(L(X'X)^{-1}L')^{-1}(Lb)
\]

where \(b = (X'X)^{-1}X'y\).

For testable hypotheses in a multivariate model, the usual multivariate tests are defined by using

\[
H = M'(LB)'(L(X'X)^{-1}L')^{-1}(LB)M
\]

where \(B = (X'X)^{-1}X'Y\) and \(Y\) is the matrix of multivariate responses or dependent variables.

The degrees of freedom associated with the hypothesis are equal to the row rank of L. The sum of squares computed in this situation is equivalent to the sum of squares computed using an L matrix with any row deleted that is a linear combination of previous rows.

Multiple-degrees-of-freedom hypotheses can be specified by separating the rows of the L matrix with commas.

---

**MANOVA Statement**

```
MANOVA 'label' < test-options > < / detail-options > ;
```

If the MODEL statement includes more than one dependent variable, you can indicate a multivariate model and define transformations of dependent variables by using the MANOVA statement.

The MANOVA statement enables you to define custom Type III hypothesis tests by specifying an M vector or matrix for testing the hypothesis \(L\beta M = 0\). The L matrix consists of one or more between-subject contrasts that involve the model effects, and the M matrix consists of one or more within-subject contrasts.

To use this feature, you must be familiar with the details of multivariate model and contrast parameterizations that are used in PROC GLM. For more information, see the sections “Multivariate Analysis of Variance” on page 3812 and “Repeated Measures Analysis of Variance” on page 3814 in Chapter 48, “The GLM Procedure.” For information about the power and sample size computational methods and formulas, see the section “Contrasts in Fixed-Effect Multivariate Models” on page 3953.

You can use either the MANOVA statement or the REPEATED statement with any of the tests for multivariate models that are supported in the MTEST= option in the POWER statement. For handling repeated measures on the same experimental unit, you would usually use the REPEATED statement instead of the MANOVA statement. But you can use the MANOVA statement in repeated measures situations, in addition to situations where you have clusters or multiple outcome variables. The differences between the MANOVA and REPEATED statements are as follows:
You can use the MANOVA statement to construct any M matrix, but you must specify the coefficients explicitly (except for the default identity matrix).

You can use the REPEATED statement to specify commonly used contrasts by using keywords rather than coefficients, but you are limited to only those forms of the M matrix.

There is no limit to the number of MANOVA statements that you can specify. Each power analysis includes tests for all MANOVA statements.

The label identifies the dependent variable transformation in the output. The label serves the same purpose as the factor-name in the REPEATED statement, enabling you to use the MANOVA statement for tests of within-subject effects and within-subject-by-between-subject interactions. A label is required for every transformation that is specified. Labels must be enclosed in single or double quotation marks.

**Test Options**

You can specify the following test-option in the MANOVA statement:

M=equation, . . . , equation | (row-of-matrix, . . . , row-of-matrix)

specifies a transformation matrix for the dependent variables that are listed in the MODEL statement.

The equations in the M= specification are of the form

\[ c_1 \times \text{dependent-variable} \pm c_2 \times \text{dependent-variable} \]

\[ \cdots \pm c_n \times \text{dependent-variable} \]

where the \( c_i \) values are coefficients of the various dependent-variables. If the value of a given \( c_i \) is 1, it can be omitted; in other words, \( 1 \times Y \) is the same as \( Y \). Equations should involve two or more dependent variables.

Alternatively, you can input the transformation matrix directly by entering the elements of the matrix, using commas to separate the rows and parentheses to surround the matrix. When you use this alternate form of input, the number of elements in each row must equal the number of dependent variables. Although these combinations actually represent the columns of the M matrix, they are displayed by rows.

When you include an M= specification, the tests are based on the variables that are defined by the equations in the specification, not the original dependent variables. If you omit the M= option, the tests are based on the original dependent variables in the MODEL statement. Omitting the M= option is equivalent to specifying an identity matrix, as in the following example, which assumes three dependent variables:

```
MANOVA 'Identity' M=(1 0 0,
                  0 1 0,
                  0 0 1);
```

For more examples of the M= option, see the section “Examples” on page 3751 in Chapter 48, “The GLM Procedure.” The syntax and functionality of the M= option in PROC GLM are the same as in PROC GLMPOWER.
**Detail Options**

You can specify the following `detail-option` in the `MANOVA` statement after a slash (/):

**ORTH**

requests that the transformation matrix in the `M=` specification of the `MANOVA` statement be orthogonalized by rows before the analysis.

---

**MODEL Statement**

```plaintext
MODEL dependent-variables = independent-effects;
```

The `MODEL` statement names the dependent variables and independent effects. If one or more `MANOVA` or `REPEATED` statements are specified, then multiple dependent variables define a multivariate model. In the absence of the `MANOVA` and `REPEATED` statements, a univariate model is assumed, and multiple dependent variables represent different scenarios for the cell means.

The `independent-effects` can involve classification variables, continuous variables, or both. You can include main effects and interactions by using the effects notation of PROC GLM; for more information, see the section “Specification of Effects” on page 3773 in Chapter 48, “The GLM Procedure.” For any model effect that involves classification variables (main effects and interactions), the number of levels cannot exceed 32,767. If no independent effects are specified, only an intercept term is fit. You can specify only one `MODEL` statement, and it must appear before the `POWER` statement if the `EFFECTS=` option is specified in the `POWER` statement.

For a univariate model, you can account for covariates without specifying them explicitly in the model by using the `NCOVARIATES=` option and either the `CORRXY=` or `PROPVARREDUCTION=` option in the `POWER` statement. For a multivariate model, you must explicitly specify any covariates in the `MODEL` statement.

The values of dependent variables in the exemplary data set (the data set named by the `DATA=` option in the `PROC GLMPOWER` statement) are surmised response means across subject profiles. For a univariate model, multiple dependent variables correspond to multiple scenarios for these cell means.

The `MODEL` statement is required. You can specify only one `MODEL` statement.

---

**PLOT Statement**

```plaintext
PLOT <plot-options> </graph-options>;
```

The `PLOT` statement produces a graph or set of graphs for the sample size analysis defined by the previous `POWER` statement. The `plot-options` define the plot characteristics, and the `graph-options` are like those in SAS/GRAPH software. If ODS Graphics is enabled, then the `PLOT` statement uses ODS Graphics to create graphs. For example:
ods graphics on;

proc glmpower data=Exemplary;
  class Variety Exposure;
  model Height = Variety | Exposure;
  power
    stddev = 4 6.5
    ntotal = 60
    power = .;
  plot x=n min=30 max=90;
run;

ods graphics off;

Otherwise, traditional graphics are produced. For example:

ods graphics off;

proc glmpower data=Exemplary;
  class Variety Exposure;
  model Height = Variety | Exposure;
  power
    stddev = 4 6.5
    ntotal = 60
    power = .;
  plot x=n min=30 max=90;
run;

For more information about enabling and disabling ODS Graphics, see the section “Enabling and Disabling ODS Graphics” on page 615 in Chapter 21, “Statistical Graphics Using ODS.”

Table 50.4 summarizes the options available in the PLOT statement.

<table>
<thead>
<tr>
<th>Option</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>Plot Options</strong></td>
<td></td>
</tr>
<tr>
<td>INTERPOL=</td>
<td>Specifies the type of curve to draw</td>
</tr>
<tr>
<td>KEY=</td>
<td>Specifies the style of key for the plot</td>
</tr>
<tr>
<td>MARKERS=</td>
<td>Specifies the locations for plotting symbols</td>
</tr>
<tr>
<td>MAX=</td>
<td>Specifies the maximum of the range of values</td>
</tr>
<tr>
<td>MIN=</td>
<td>Specifies the minimum of the range of values</td>
</tr>
<tr>
<td>NPOINTS=</td>
<td>Specifies the number of values</td>
</tr>
<tr>
<td>STEP=</td>
<td>Specifies the increment between values</td>
</tr>
<tr>
<td>VARY</td>
<td>Specifies how plot features should be linked to varying analysis parameters</td>
</tr>
<tr>
<td>X=</td>
<td>Specifies a plot with the requested type of parameter on the X axis</td>
</tr>
<tr>
<td>XOPTS=</td>
<td>Specifies plot characteristics pertaining to the X axis</td>
</tr>
<tr>
<td>Y=</td>
<td>Specifies a plot with the requested type of parameter on the Y axis</td>
</tr>
<tr>
<td>YOPTS=</td>
<td>Specifies plot characteristics pertaining to the Y axis</td>
</tr>
<tr>
<td><strong>Graph Options</strong></td>
<td></td>
</tr>
<tr>
<td>DESCRIPTION=</td>
<td>Specifies a descriptive string</td>
</tr>
<tr>
<td>NAME=</td>
<td>Specifies a name for the catalog entry for the plot</td>
</tr>
</tbody>
</table>

Table 50.4  PLOT Statement Options
Options

You can specify the following plot-options in the PLOT statement.

**INTERPOL=JOIN | NONE**

specifies the type of curve to draw through the computed points. The INTERPOL=JOIN option connects computed points with straight lines. The INTERPOL=NONE option leaves computed points unconnected.

**KEY=BYCURVE < (bycurve-options) >**

**KEY=BYFEATURE < (byfeature-options) >**

**KEY=ONCURVES**

specifies the style of key (or “legend”) for the plot. The default is KEY=BYFEATURE, which specifies a key with a column of entries for each plot feature (line style, color, and/or symbol). Each entry shows the mapping between a value of the feature and the value(s) of the analysis parameter(s) linked to that feature. The KEY=BYCURVE option specifies a key with each row identifying a distinct curve in the plot. The KEY=ONCURVES option places a curve-specific label adjacent to each curve.

You can specify the following byfeature-options in parentheses after the KEY=BYCURVE option.

**NUMBERS=OFF | ON**

specifies how the key should identify curves. If NUMBERS=OFF, then the key includes symbol, color, and line style samples to identify the curves. If NUMBERS=ON, then the key includes numbers matching numeric labels placed adjacent to the curves. The default is NUMBERS=ON.

**POS=BOTTOM | INSET**

specifies the position of the key. The POS=BOTTOM option places the key below the X axis. The POS=INSET option places the key inside the plotting region and attempts to choose the least crowded corner. The default is POS=BOTTOM.

You can specify the following byfeature-options in parentheses after KEY=BYFEATURE option.

**POS=BOTTOM | INSET**

specifies the position of the key. The POS=BOTTOM option places the key below the X axis. The POS=INSET option places the key inside the plotting region and attempts to choose the least crowded corner. The default is POS=BOTTOM.

**MARKERS=ANALYSIS | COMPUTED | NICE | NONE**

specifies the locations for plotting symbols.

The MARKERS=ANALYSIS option places plotting symbols at locations that correspond to the values of the relevant input parameter from the POWER statement preceding the PLOT statement.

The MARKERS=COMPUTED option (the default) places plotting symbols at the locations of actual computed points from the sample size analysis.

The MARKERS=NICE option places plotting symbols at tick mark locations (corresponding to the argument axis).

The MARKERS=NONE option disables plotting symbols.
**MAX=number | DATAMAX**
specifies the maximum of the range of values for the parameter associated with the “argument” axis (the axis that is not representing the parameter being solved for). The default is DATAMAX, which specifies the maximum value that occurs for this parameter in the POWER statement that precedes the PLOT statement.

**MIN=number | DATAMIN**
specifies the minimum of the range of values for the parameter associated with the “argument” axis (the axis that is not representing the parameter being solved for). The default is DATAMIN, which specifies the minimum value that occurs for this parameter in the POWER statement that precedes the PLOT statement.

**NPOINTS=number**

**NPTS=number**
specifies the number of values for the parameter associated with the “argument” axis (the axis that is not representing the parameter being solved for). You cannot use the NPOINTS= and STEP= options simultaneously. The default value for typical situations is 20.

**STEP=number**
specifies the increment between values of the parameter associated with the “argument” axis (the axis that is not representing the parameter being solved for). You cannot use the STEP= and NPOINTS= options simultaneously. By default, the NPOINTS= option is used instead of the STEP= option.

**VARY ( feature < BY parameter-list > < , . . . , feature < BY parameter-list > > )**
specifies how plot features should be linked to varying analysis parameters. Available features are COLOR, LINESTYLE, PANEL, and SYMBOL. A “panel” refers to a separate plot with a heading identifying the subset of values represented in the plot.

The parameter-list is a list of one or more names, separated by spaces. Each name must match the name of an analysis option used in the POWER statement preceding the PLOT statement, or one of the following keywords:

- **SOURCE**, which represents the model effects and contrasts in a univariate model and the between-subject effects and contrasts in a multivariate model
- **DEPENDENT**, which represents the cell means scenarios in a univariate model or the dependent variable transformations in a multivariate model

If the name represents an analysis option that is specified in the POWER statement, then it must be the primary name for the analysis option—that is, the one that is listed first in the syntax description.

If you omit the < BY parameter-list > portion for a feature, then one or more multivalued parameters from the analysis are automatically selected for you.

**X=N | POWER**
specifies a plot with the requested type of parameter on the X axis and the parameter being solved for on the Y axis. When X=N, sample size is assigned to the X axis. When X=POWER, power is assigned to the X axis. You cannot use the X= and Y= options simultaneously. The default is X=POWER, unless the result parameter is power, in which case the default is X=N.
**XOPTS= ( x-options )**
specifies plot characteristics pertaining to the X axis.

You can specify the following x-options in parentheses.

**CROSSREF=NO | YES**
specifies whether the reference lines defined by the REF= x-option should be crossed with a reference line on the Y axis that indicates the solution point on the curve.

**REF=number-list**
specifies locations for reference lines extending from the X axis across the entire plotting region. For information about specifying the number-list, see the section “Specifying Value Lists in the POWER Statement” on page 3946.

**Y=N | POWER**
specifies a plot with the requested type of parameter on the Y axis and the parameter being solved for on the X axis. When Y=N, sample size is assigned to the Y axis. When Y=POWER, power is assigned to the Y axis. You cannot use the Y= and X= options simultaneously. By default, the X= option is used instead of the Y= option.

**YOPTS= ( y-options )**
specifies plot characteristics pertaining to the Y axis.

You can specify the following y-options in parentheses.

**CROSSREF=NO | YES**
specifies whether the reference lines defined by the REF= y-option should be crossed with a reference line on the X axis that indicates the solution point on the curve.

**REF=number-list**
specifies locations for reference lines extending from the Y axis across the entire plotting region. For information about specifying the number-list, see the section “Specifying Value Lists in the POWER Statement” on page 3946.

You can specify the following graph-options in the PLOT statement after a slash (/).

**DESCRIPTION='string’**
specifies a descriptive string of up to 40 characters that appears in the “Description” field of the graphics catalog. The description does not appear on the plots. By default, PROC GLMPOWER assigns a description either of the form “Y versus X” (for a single-panel plot) or of the form “Y versus X (S),” where Y is the parameter on the Y axis, X is the parameter on the X axis, and S is a description of the subset represented on the current panel of a multipanel plot.

**NAME='string’**
specifies a name of up to eight characters for the catalog entry for the plot. The default name is PLOTn, where n is the number of the plot statement within the current invocation of PROC GLMPOWER. If the name duplicates the name of an existing entry, SAS/GRAPH software adds a number to the duplicate name to create a unique entry—for example, PLOT11 and PLOT12 for the second and third panels of a multipanel plot generated in the first PLOT statement in an invocation of PROC GLMPOWER.
Chapter 50: The GLMPOWER Procedure

POWER Statement

\[
\text{POWER <options> ;}
\]

The POWER statement performs power and sample size analyses for the Type III \(F\) tests that are specified in the MTEST= option, for each effect in the model that is defined by the MODEL statement, and for the contrasts that are defined by all CONTRAST, MANOVA, and REPEATED statements. The POWER statement must appear after the MODEL statement if the EFFECTS= option is used in the POWER statement.

For information about the power and sample size computational methods and formulas, see the section “Computational Methods and Formulas” on page 3950.

Summary of Options

Table 50.5 summarizes the options available in the POWER statement.

<table>
<thead>
<tr>
<th>Option</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>Specify test statistic</td>
<td></td>
</tr>
<tr>
<td>MTEST=</td>
<td>Specifies the test statistic for a multivariate model</td>
</tr>
<tr>
<td>UEPSDEF=</td>
<td>Specifies the form of the Huynh-Feldt epsilon for MTEST=HF</td>
</tr>
<tr>
<td>Specify analysis information</td>
<td></td>
</tr>
<tr>
<td>ALPHA=</td>
<td>Specifies the level of significance of each test</td>
</tr>
<tr>
<td>EFFECTS=</td>
<td>Specifies the model effects</td>
</tr>
<tr>
<td>Specify covariates for a univariate model</td>
<td></td>
</tr>
<tr>
<td>CORRXY=</td>
<td>Specifies multiple correlation ((\rho)) between covariates and response</td>
</tr>
<tr>
<td>NCOVARIATES=</td>
<td>Specifies additional degrees of freedom due to covariates</td>
</tr>
<tr>
<td>PROPVARREDUCTION=</td>
<td>Specifies proportional variance reduction ((r)) due to covariates</td>
</tr>
<tr>
<td>Specify variability</td>
<td></td>
</tr>
<tr>
<td>CORRMAT=</td>
<td>Specifies the correlation matrix of the dependent variables in a multivariate model</td>
</tr>
<tr>
<td>CORRS=</td>
<td>Specifies the correlations among the dependent variables in a multivariate model</td>
</tr>
<tr>
<td>COVMAT=</td>
<td>Specifies the covariance matrix of the dependent variables in a multivariate model</td>
</tr>
<tr>
<td>MATRIX=</td>
<td>Defines a matrix or vector</td>
</tr>
<tr>
<td>SQRTVAR=</td>
<td>Specifies the vector of error standard deviations for each dependent variable in a multivariate model</td>
</tr>
<tr>
<td>STDDEV=</td>
<td>Specifies the common error standard deviation</td>
</tr>
<tr>
<td>Specify sample size</td>
<td></td>
</tr>
<tr>
<td>NFRACTIONAL</td>
<td>Enables fractional input and output for sample sizes</td>
</tr>
<tr>
<td>NTOTAL=</td>
<td>Specifies the sample size</td>
</tr>
<tr>
<td>Specify power</td>
<td></td>
</tr>
<tr>
<td>POWER=</td>
<td>Specifies power</td>
</tr>
<tr>
<td>Choose computational method</td>
<td></td>
</tr>
<tr>
<td>METHOD=</td>
<td>Specifies the computational method for multivariate tests</td>
</tr>
</tbody>
</table>
Table 50.5  continued

<table>
<thead>
<tr>
<th>Option</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>Control ordering in output</td>
<td></td>
</tr>
<tr>
<td>DEPENDENT</td>
<td>Specifies the location of the Dependent or Transformation column in the output</td>
</tr>
<tr>
<td>OUTPUTORDER=</td>
<td>Controls ordering in output</td>
</tr>
</tbody>
</table>

Table 50.6 summarizes the valid result parameters.

Table 50.6  Summary of Result Parameters in the POWER Statement

<table>
<thead>
<tr>
<th>Solve for</th>
<th>Syntax</th>
</tr>
</thead>
<tbody>
<tr>
<td>Power</td>
<td>POWER = .</td>
</tr>
<tr>
<td>Sample size</td>
<td>NTOTAL = .</td>
</tr>
</tbody>
</table>

Dictionary of Options

**ALPHA=number-list**

specifies the level of significance of each test. The default is 0.05, which corresponds to the usual $0.05 \times 100\% = 5\%$ level of significance. Note that this is a test-wise significance level with the same value for all tests, not incorporating any corrections for multiple testing. For information about specifying the *number-list*, see the section “Specifying Value Lists in the POWER Statement” on page 3946.

**CORRMAT= name-list**

specifies the correlation matrix of the dependent variables in a multivariate model, by using labels that are specified in the `MATRIX=` option. The corresponding matrices that are defined in the `MATRIX=` option must have either a lower triangular form that includes the diagonal of 1’s or a linear exponent autoregressive (LEAR) correlation structure. The matrix must be positive definite. You can use the `CORRMAT=` option only when you have a multivariate model—that is, in the presence of one or more `MANOVA` or `REPEATED` statements. For information about specifying the *name-list*, see the section “Specifying Value Lists in the POWER Statement” on page 3946.

**CORRS= name-list**

specifies the correlations among the dependent variables in a multivariate model, by using labels that are specified in the `MATRIX=` option. The corresponding matrices that are defined in the `MATRIX=` option must have either a lower triangular form that excludes the diagonal of 1’s or a LEAR correlation structure. The matrix must be positive definite. You can use the `CORRS=` option only when you have a multivariate model—that is, in the presence of one or more `MANOVA` or `REPEATED` statements. For information about specifying the *name-list*, see the section “Specifying Value Lists in the POWER Statement” on page 3946.
COVMAT=name-list
  specifies the covariance matrix of the dependent variables in a multivariate model, by using labels that are specified in the MATRIX= option. The corresponding matrices that are defined in the MATRIX= option must have a lower triangular form that includes the diagonal of error variances. The matrix must be positive definite. You can use the COVMAT= option only when you have a multivariate model—that is, in the presence of one or more MANOVA or REPEATED statements. For information about specifying the name-list, see the section “Specifying Value Lists in the POWER Statement” on page 3946.

CORRXY=number-list
  specifies the multiple correlation ($\rho$) between all covariates and the response for a univariate model. The error standard deviation that is given by the STDDEV= option is consequently reduced by multiplying it by a factor of $(1 - \rho^2)^{1/2}$, provided that the number of covariates (as determined by the NCOVARIATES= option) is greater than 0. You cannot use the CORRXY= and PROPARREDUCTION= options simultaneously. You cannot use the CORRXY= option when you have a multivariate model—that is, in the presence of a MANOVA or REPEATED statement. For information about specifying the number-list, see the section “Specifying Value Lists in the POWER Statement” on page 3946.

DEPENDENT
  specifies the location of the Dependent column (for a univariate model) or the Transformation column (for a multivariate model) in the output when you specify the OUTPUTORDER=REVERSE option or OUTPUTORDER=SYNTAX option, according to its relative position in the POWER statement.

EFFECTS <= < ( effect . . . effect ) >>
  specifies the model effects to include in the power analysis. By default, or if the EFFECTS keyword is specified without the equal sign (=), all model effects are included. Specify EFFECTS=() to exclude all model effect tests from the power analysis. You can include main effects and interactions by using the effects notation of PROC GLM; for more information, see the section “Specification of Effects” on page 3773 in Chapter 48, “The GLM Procedure.” The MODEL statement must appear before the POWER statement if the EFFECTS option is used.

MATRIX('label')=matrix-specification
  defines a matrix or vector that you can use along with the CORRMAT=, CORRS=, COVMAT=, and SQRTAR= options when you have a multivariate model.

  The matrix-specification can have one of the following three forms:

  1. Raw values

     (values )

     specifies the values of a matrix or vector in one of the following forms:

     - a matrix in lower triangular form, for use with the CORRMAT= or COVMAT= option
     - a matrix in strictly lower triangular form, for use with the CORRS= option
     - a vector, for use with the SQRTAR= option

     A matrix in lower triangular form contains the diagonal the and values below it. For example, you can represent a $3 \times 3$ correlation matrix for use with the CORRMAT= option as follows:
You can represent the same correlation matrix in strictly lower triangular form for use with the CORRS= option as follows:

\[
\text{MATRIX ('MyCorrs') = (0.5 } \\
\hspace{1cm} 0.2 0.5) \\
\]

An example of a vector for use with the SQRTVAR= option is as follows:

\[
\text{MATRIX ('MySqrtVar') = (3.2 } \\
\hspace{1cm} 4.5 3.7) \\
\]

2. Linear exponent autoregressive (LEAR) correlation structure

\[
\text{LEAR (base-corr, corr-decay <, nlevels <, level-values >>) }
\]

specifies a LEAR correlation structure for use with the CORRMAT= or CORRS= option. The LEAR structure is useful for characterizing exponentially decaying within-subject correlation that decays at a rate slower or faster than AR(1). Special cases include compound symmetry, first-order autoregressive (AR(1)), and first-order moving average correlation structures.

The LEAR correlation structure is related to the spatial covariance structures in PROC MIXED and is discussed in Simpson et al. (2010).

The base-corr (\(\rho\)) is the correlation between variables whose level-values are one unit apart, and it must satisfy \(0 \leq \rho < 1\). The corr-decay (\(\delta\)) is the correlation decay rate, which must be nonnegative. The default value for the number of levels, nlevels (\(n\)), is the number of dependent variables. The n level-values, denoted \(\{l_1, \ldots, l_n\}\), must be distinct. The default level-values are \(\{1, \ldots, n\}\). Let \(d_{jk}\) denote the distance between levels \(j\) and \(k\), \(d_{jk} = |l_j - l_k|\). Let \(d_{\min} = \min(d_{jk} : j \neq k)\) and \(d_{\max} = \max(d_{jk} : j \neq k)\). The \((i, j)\) element of the correlation matrix according to the LEAR model is defined as

\[
\rho_{jk} = \begin{cases} 
1 & \text{if } j = k \\
\rho^{d_{\min} + \delta[(d_{jk} - d_{\min})/(d_{\max} - d_{\min})]} & \text{if } j \neq k \text{ and } d_{\min} \neq d_{\max} \\
\rho & \text{if } j \neq k \text{ and } d_{\min} = d_{\max}
\end{cases}
\]

Compound symmetry is the special case \(\delta = 0\). AR(1) is the special case \(\delta = d_{\max} - d_{\min}\). As \(\delta \to \infty\), the model approaches the first-order moving average model.

3. Kronecker product

\[
\text{‘matrix-name’ @ ‘matrix-name’ @ … @ ‘matrix-name’}
\]

specifies a direct (Kronecker) product of two or more matrices for use with the CORRMAT=, CORRS=, or COVMAT= option. This form is useful when you have more than one type of distinction among dependent variables. For example, suppose you have a three-level repeated measurement factor with correlation that is 0.4 for neighboring measurements and that decays slightly more slowly than AR(1)
across more distant measurements. You also have four clusters that you believe satisfy compound symmetry with a correlation of 0.3. Your level values are the same as the default for the LEAR model. You can specify this correlation structure as follows:

\[
\text{MATRIX ('RepMeasures') = LEAR (0.4, 1.5, 3)}
\]
\[
\text{MATRIX ('Clusters') = LEAR (0.3, 0, 4)}
\]
\[
\text{MATRIX ('FullCorr') = 'RepMeasures' @ 'Clusters'}
\]
\[
\text{CORRMAT = 'FullCorr'}
\]

You can use the \text{MATRIX=} option only when you have a multivariate model—that is, in the presence of one or more \text{MANOVA} or \text{REPEATED} statements.

\text{METHOD=MULLERPETERSON | MP}
\text{METHOD=OBRIENSHIEH | OS}

specifies the power computation method for the multivariate tests (\text{MTEST=HLT}, \text{MTEST=PT}, and \text{MTEST=WILKS}). \text{METHOD=OBRIENSHIEH} (the default) is based on O'Brien and Shieh (1992), and \text{METHOD=MULLERPETERSON} is based on Muller and Peterson (1984). For information about the associated power and sample size computational methods and formulas, see the section “Multivariate Tests” on page 3955.

If the dependent transformation consists of a single contrast \((r_M = 1)\), then the two methods are identical and compute exact power. If \(r_M > 1\) but the model effect or between-subject contrast has only one degree of freedom \((r_L = 1)\), then \text{METHOD=OBRIENSHIEH} computes exact results and \text{METHOD=MULLERPETERSON} computes approximate results. If \(r_M > 1\) and \(r_L > 1\), then both methods compute approximate results.

You can use the \text{METHOD=} option only when you have a multivariate model—that is, in the presence of one or more \text{MANOVA} or \text{REPEATED} statements.

\text{MTEST=}\text{test-list}

specifies the form of the \(F\) test for a multivariate model. Seven keywords are available, as discussed in the following paragraphs: \text{BOX}, \text{GG}, \text{HF}, \text{HLT}, \text{PT}, \text{UNCORR}, and \text{WILKS}. For information about specifying the \text{keyword-list}, see the section “Specifying Value Lists in the \text{POWER} Statement” on page 3946.

Three of these tests are multivariate, corresponding to the default \text{MSTAT=FAPPROX} option in the \text{MANOVA} and \text{REPEATED} statements in PROC GLM:

- \text{MTEST=HLT} (the default) is the Hotelling-Lawley trace
- \text{MTEST=PT} is Pillai’s trace
- \text{MTEST=WILKS} is Wilks’ lambda

For more information about these multivariate tests, see the section “Multivariate Tests” on page 94 in Chapter 4, “Introduction to Regression Procedures.” For information about the associated power and sample size computational methods and formulas, see the section “Multivariate Tests” on page 3955.

The other four tests are univariate, corresponding to the univariate approach to repeated measures in the \text{REPEATED} statement in PROC GLM:

- \text{MTEST=UNCORR} is the uncorrected univariate \(F\) test, assuming sphericity \((\varepsilon = 1)\).
• **MTEST=GG** is the $F$ test with the Greenhouse-Geisser adjustment, estimating $\varepsilon$ by its maximum likelihood estimate $\hat{\varepsilon}$.

• **MTEST=HF** is the $F$ test with the Huynh-Feldt adjustment as specified by the UEPSDEF= option, using an approximately unbiased estimate $\hat{\varepsilon}$.

• **MTEST=BOX** is the $F$ test with Box’s conservative adjustment, estimating sphericity by its smallest possible value $1/r_M$, where $r_M$ is the number of within-subject contrasts in the dependent variable transformation.

For more information about these univariate tests, see the section “Hypothesis Testing in Repeated Measures Analysis” on page 3816 in Chapter 48, “The GLM Procedure.” For information about the associated power and sample size computational methods and formulas, see the section “Univariate Tests” on page 3958.

These tests are all of the form $L\beta M = 0$, where $L$ is a between-subject contrast, $\beta$ is the matrix of model parameters, and $M$ is a within-subject contrast.

You can use the **MTEST=** option only when you have a multivariate model—that is, in the presence of one or more **MANOVA** or **REPEATED** statements.

**NCOVARIATES=number-list**

**NCOVARIATE=number-list**

**NCOVS=number-list**

**NCOV=number-list**

specifies the number of additional degrees of freedom to accommodate covariate effects—both class and continuous—not listed in the **MODEL** statement, for a univariate model. The error degrees of freedom are consequently reduced by the value of the **NCOVARIATES=** option, and the error standard deviation (whose unadjusted value is provided with the STDDEV= option) is reduced according to the value of the **CORRXY=** or **PROPVARREDUCTION=** option. You cannot use the **NCOVARIATES=** option when you have a multivariate model—that is, in the presence of a **MANOVA** or **REPEATED** statement. For information about specifying the **number-list**, see the section “Specifying Value Lists in the POWER Statement” on page 3946.

**NFRATIONAL**

**NFRAC**

enables fractional input and output for sample sizes. See the section “Sample Size Adjustment Options” on page 3947 for information about the ramifications of the presence (and absence) of the **NFRATIONAL** option.

**NTOTAL=number-list**

specifies the sample size or requests a solution for the sample size by specifying a missing value (**NTOTAL=**). The term “sample size” here refers to the number of independent sampling units. Values for the sample size for a univariate model must be no smaller than the model degrees of freedom (counting the covariates if present). The minimum required sample size for a multivariate model depends on the analysis and computational method; for more information, see the section “Contrasts in Fixed-Effect Multivariate Models” on page 3953. For information about specifying the **number-list**, see the section “Specifying Value Lists in the POWER Statement” on page 3946.
**OUTPUTORDER=INTERNAL | REVERSE | SYNTAX**

controls how the input and default analysis parameters are ordered in the output. **OUTPUTORDER=INTERNAL** (the default) arranges the parameters in the output according to the following order of their corresponding options:

- DEPENDENT
- EFFECTS=
  - weight variable (from the WEIGHT statement)
- ALPHA=
- NCOVARIATES=
- CORRXY=
- PROPVARREDUCTION=
- STDDEV=
- NTOTAL=
- POWER=

The **OUTPUTORDER=SYNTAX** option arranges the parameters in the output in the same order in which their corresponding options are specified in the POWER statement. The **OUTPUTORDER=REVERSE** option arranges the parameters in the output in the reverse of the order in which their corresponding options are specified in the POWER statement.

**POWER=number-list**

specifies the desired power of each test or requests a solution for the power by specifying a missing value (POWER=.). The power is expressed as a probability (for example, 0.9) rather than a percentage. Note that this is a test-wise power with the same value for all tests, without any correction for multiple testing. For information about specifying the number-list, see the section “Specifying Value Lists in the POWER Statement” on page 3946.

**PROPVARREDUCTION=number-list**

**PVRED=number-list**

specifies the proportional reduction \( r \) in total R square incurred by the covariates—in other words, the amount of additional variation explained by the covariates—for a univariate model. The error standard deviation that is given by the STDDEV= option is consequently reduced by multiplying it by a factor of \( (1 - r)^{\frac{1}{2}} \), provided that the number of covariates (as determined by the NCOVARIATES= option) is greater than 0. You cannot use the PROPVARREDUCTION= and CORRXY= options simultaneously. You cannot use the PROPVARREDUCTION= option when you have a multivariate model—that is, in the presence of a MANOVA or REPEATED statement. For information about specifying the number-list, see the section “Specifying Value Lists in the POWER Statement” on page 3946.

**SQRTVAR=name-list**

specifies the vector of standard deviations—that is, the square roots of the variances—of the dependent variables in a multivariate model, by using labels that are specified using the MATRIX= option. The standard deviation values must be positive. You can use the SQRTVAR= option only when you have a multivariate model—that is, in the presence of one or more MANOVA or REPEATED statements. For information about specifying the name-list, see the section “Specifying Value Lists in the POWER Statement” on page 3946.
STDDEV=number-list

specifies the error standard deviation, or root MSE. For a multivariate model, each value in the number-list is taken to be a common value for all dependent variables. If covariates are specified by using the NCOVARIATES= option, then the STDDEV= option denotes the error standard deviation before accounting for these covariates. For information about specifying the number-list, see the section “Specifying Value Lists in the POWER Statement” on page 3946.

UEPSDEF=unbiased-epsilon-definition

specifies the type of adjustment for MTEST=HF. The default is UEPSDEF=HFL, which corresponds to the corrected form of the Huynh-Feldt adjustment (Huynh and Feldt 1976; Lecoutre 1991). Other alternatives are UEPSDEF=HF; the uncorrected Huynh-Feldt adjustment; and UEPSDEF=CM, the adjustment of Chi et al. (2012). For more information about these adjustments, see the section “Hypothesis Testing in Repeated Measures Analysis” on page 3816 in Chapter 48, “The GLM Procedure.” You can use the UEPSDEF= option only when you have a multivariate model—that is, in the presence of one or more MANOVA or REPEATED statements. For information about the associated power and sample size computational methods and formulas, see the section “Univariate Tests” on page 3958.

Restrictions on Option Combinations

To specify the variability in a multivariate model, choose one of the following parameterizations:

- covariance matrix (using the MATRIX= and COVMAT= options)
- standard deviations and correlations (using the MATRIX=, SQRTVAR=, and CORRS= options)
- common standard deviation and correlations (using the STDDEV=, MATRIX=, and CORRS= options)
- standard deviations and correlation matrix (using the MATRIX=, SQRTVAR=, and CORRMAT= options)
- common standard deviation and correlation matrix (using the STDDEV=, MATRIX=, and CORRMAT= options)

For the relationship between covariates and response in a univariate model, specify either the multiple correlation (by using the CORRXY= option) or the proportional reduction in total R square (by using the PROPVARREDUCTION= option).

**REPEATED Statement**

REPEATED factor-specification ;

If the MODEL statement includes more than one dependent variable, you can indicate a multivariate model and define transformations of dependent variables by using the REPEATED statement.

The REPEATED statement enables you to define custom Type III hypothesis tests by choosing from among several transformations of the dependent variables: contrast, Helmert, identity, mean, polynomial, and profile. You can specify a transformation for each repeated factor (often called a within-subject factor), and each combination of repeated factors produces an M vector or matrix for testing the hypothesis $L\beta M = 0$. The L matrix consists of one or more between-subject contrasts that involve the model effects, and the M matrix
Chapter 50: The GLMPOWER Procedure

consists of one or more within-subject contrasts that involve the repeated factors. There is no limit to the number of repeated factors that you can specify.

Usually, the variables on the left side of the equation in the MODEL statement represent one repeated response variable. This does not mean that you are limited to listing only one factor in the REPEATED statement. For example, one repeated response variable (wellness rating) might be measured six times (implying variables Y1 to Y6 on the left side of the equal sign in the MODEL statement), with the associated within-subject factors rater and time (implying two factors listed in the REPEATED statement). However, designs that have two or more repeated response variables can be handled by using the IDENTITY transformation.

To use this feature, you must be familiar with the details of multivariate model and contrast parameterizations that PROC GLM uses. For more information, see the sections “Repeated Measures Analysis of Variance” on page 3814 and “Multivariate Analysis of Variance” on page 3812 in Chapter 48, “The GLM Procedure.” For information about the power and sample size computational methods and formulas, see the section “Contrasts in Fixed-Effect Multivariate Models” on page 3953.

If you specify one or more REPEATED statements, then a “Mean(Dep)” transformation is added to the power analysis. This transformation is the mean of the dependent variables, the same transformation that is used implicitly in the “Tests of Hypotheses for Between Subjects Effects” table in PROC GLM. In addition, the Intercept model effect is included in the power analysis. If the REPEATED statement is not specified, then tests that involve the Intercept are excluded from the power analysis.

You can use either the REPEATED statement or the MANOVA statement along with any of the tests for multivariate models that are supported in the MTEST= option in the POWER statement. The REPEATED statement is usually used for handling repeated measurements on the same experimental unit, but you can also use the REPEATED statement for other situations, such as clusters or multiple outcome variables. The differences between the REPEATED and MANOVA statements are as follows:

- You can use the REPEATED statement to specify commonly used contrasts by using keywords rather than coefficients, but you are limited to only those forms of the M matrix.
- You can use the MANOVA statement to construct any M matrix, but you must specify the coefficients explicitly (except for the default identity matrix).

There is no limit to the number of REPEATED statements that you can specify. Each power analysis includes tests for all REPEATED statements and also (if you specify at least one REPEATED statement) the extra “Mean(Dep)” transformation that was previously mentioned.

The simplest form of the REPEATED statement requires only a factor-name. When you have two or more repeated factors, you must specify the factor-name and number of levels (levels) for each factor. Optionally, you can specify the actual values for the levels (level-values) and a transformation that defines single-degree-of-freedom contrasts. When you specify more than one within-subject factor, the factor-names (and associated level and transformation information) must be separated by a comma in the REPEATED statement.

The factor-specification for the REPEATED statement can include any number of individual factor specifications, separated by commas, of the following form:

\[ \text{factor-name levels} < \text{(level-values)}> <\text{transformation}> \]
where

**factor-name** names a factor to be associated with the dependent variables. The name should not be the same as any variable name that already exists in the data set being analyzed and should conform to the usual conventions of SAS variable names.

When you specify more than one factor, list the dependent variables in the **MODEL** statement so that the within-subject factors that you define in the **REPEATED** statement are nested; that is, the first factor you define in the **REPEATED** statement should be the one whose values change least frequently.

**levels** gives the number of levels associated with the factor being defined. When there is only one within-subject factor, the number of levels is equal to the number of dependent variables. In this case, **levels** is optional. When more than one within-subject factor is defined, however, **levels** is required, and the product of the number of levels of all the factors must equal the number of dependent variables in the **MODEL** statement.

**(level-values)** gives values that correspond to levels of a repeated measures factor. These values are used as spacings for constructing orthogonal polynomial contrasts if you specify a POLYNOMIAL transformation. The number of values that you specify must correspond to the number of levels for that factor in the **REPEATED** statement. Enclose the **level-values** in parentheses.

The following transformation keywords define single-degree-of-freedom contrasts for factors that you specify in the **REPEATED** statement. Because the number of contrasts that are generated is always one less than the number of levels of the factor, you have some control over which contrast is omitted from the analysis by which transformation you select. The only exception is the IDENTITY transformation, which is not composed of contrasts and has the same degrees of freedom as the factor has levels. By default, PROC GLMPOWER uses the CONTRAST transformation.

**CONTRAST**(ordinal-reference-level)>

generates contrasts between levels of the factor and a reference level. By default, PROC GLMPOWER uses the last level as the reference level; you can optionally specify a reference level in parentheses after the keyword CONTRAST. The reference level corresponds to the ordinal value of the level rather than the level value that is specified. For example, to generate contrasts between the first level of a factor and the other levels, specify CONTRAST(1).

**HELMERT**

generates contrasts between each level of the factor and the mean of subsequent levels.

**IDENTITY**

generates an identity transformation that corresponds to the associated factor. This transformation is not composed of contrasts; it has \( n \) degrees of freedom for an \( n \)-level factor, instead of \( n - 1 \) degrees of freedom.

**MEAN**(ordinal-reference-level)>

generates contrasts between levels of the factor and the mean of all other levels of the factor. Specifying a reference level eliminates the contrast between that level and the mean. When no reference level is specified, the contrast that involves the last level is omitted. For an example, see the CONTRAST transformation.
Chapter 50: The GLMPOWER Procedure

**POLYNOMIAL**

generates orthogonal polynomial contrasts. Level values, if provided, are used as spacings in the construction of the polynomials; otherwise, equal spacing is assumed.

**PROFILE**

generates contrasts between adjacent levels of the factor.

**Examples**

When you specify more than one factor, list the dependent variables in the **MODEL** statement so that the within-subject factors that you define in the **REPEATED** statement are nested; that is, the first factor you define in the **REPEATED** statement should be the one whose values change least frequently. For example, assume that two raters submit a wellness rating at each of three times, for a total of six dependent variables for each subject. Consider the following statements:

```
proc glm;
  class treatment;
  model Y1-Y6 = treatment;
  repeated rater 2, time 3;
run;
```

The variables are listed in the **MODEL** statement as Y1 through Y6, so the **REPEATED** statement in the preceding statements implies the following structure:

<table>
<thead>
<tr>
<th>Dependent Variables</th>
<th>Y1</th>
<th>Y2</th>
<th>Y3</th>
<th>Y4</th>
<th>Y5</th>
<th>Y6</th>
</tr>
</thead>
<tbody>
<tr>
<td>Value of rater</td>
<td>1</td>
<td>1</td>
<td>1</td>
<td>2</td>
<td>2</td>
<td>2</td>
</tr>
<tr>
<td>Value of time</td>
<td>1</td>
<td>2</td>
<td>3</td>
<td>1</td>
<td>2</td>
<td>3</td>
</tr>
</tbody>
</table>

**WEIGHT Statement**

```
WEIGHT variable;
```

The **WEIGHT** statement names a variable that provides a profile weight (“cell weight”) for each observation in the exemplary data set specified by the **DATA=** option in the **PROC GLMPOWER** statement.

If the **WEIGHT** statement is not used, then a balanced design is assumed with default cell weights of 1.

**Details: GLMPOWER Procedure**

**Specifying Value Lists in the POWER Statement**

To specify one or more scenarios for an analysis parameter (or set of parameters) in the **POWER** statement, you provide a list of values for the option that corresponds to the parameter(s). To identify the parameter you want to solve for, you place a missing value in the appropriate list.
There are three basic types of such lists: *number-lists*, *name-lists*, and *keyword-lists*. Scenarios for scalar-valued parameters, such as power, are represented by a *number-list*. Scenarios for named parameters, such as correlation matrices, are represented by a *name-list*. Some parameters, such as the test statistic for a multivariate model, have values that are represented by one or more keywords in a *keyword-list*.

**Number-Lists**

A *number-list* can be one of two things: a series of one or more numbers expressed in the form of one or more DOLISTs, or a missing value indicator ( . ).

The DOLIST format is the same as in the DATA step. For example, you can specify four scenarios (30, 50, 70, and 100) for a total sample size in either of the following ways:

\[
\begin{align*}
\text{NTOTAL} &= 30 \ 50 \ 70 \ 100 \\
\text{NTOTAL} &= 30 \text{ to } 70 \text{ by } 20 \ 100
\end{align*}
\]

A missing value identifies a parameter as the result parameter; it is valid only with options representing parameters you can solve for in a given analysis. For example, you can request a solution for NTOTAL as follows:

\[
\text{NTOTAL} = .
\]

**Name-Lists**

A *name-list* is a list of one or more names that are enclosed in single or double quotation marks and separated by spaces. For example, you can specify two scenarios for the correlation matrix in a multivariate model as follows:

\[
\text{CORRMAT} = "\text{Corr A}" \ "\text{Corr B}"
\]

**Keyword-Lists**

A *keyword-list* is a list of one or more keywords, separated by spaces. For example, you can specify both the multivariate Hotelling-Lawley trace and uncorrected univariate \(F\) test for a multivariate model as follows:

\[
\text{MTEST} = \text{HLT UNCORR}
\]

**Sample Size Adjustment Options**

By default, PROC GLMPOWER rounds sample sizes conservatively (down in the input, up in the output) so that all total sizes and sample sizes for individual design profiles are integers. This is generally considered conservative because it selects the closest realistic design providing at most the power of the (possibly fractional) input or mathematically optimized design. In addition, all design profile sizes are adjusted to be multiples of their corresponding weights. If a design profile is present more than once in the exemplary data set, then the weights for that design profile are summed. For example, if a particular design profile is present twice in the exemplary data set with weight values 2 and 6, then all sample sizes for this design profile become multiples of \(2 + 6 = 8\).
With the **NFRACTIONAL** option, sample size input is not rounded, and sample size output is reported in two versions, a raw “fractional” version and a “ceiling” version rounded up to the nearest integer.

Whenever an input sample size is adjusted, both the original (“nominal”) and adjusted (“actual”) sample sizes are reported. Whenever computed output sample sizes are adjusted, both the original input (“nominal”) power and the achieved (“actual”) power at the adjusted sample size are reported.

## Error and Information Output

The Error column in the main output table explains reasons for missing results and flags numerical results that are bounds rather than exact answers.

The Info column provides further information about Error entries, warnings about any boundary conditions detected, and notes about any adjustments to input. Note that the Info column is hidden by default in the main output. You can view it by using the ODS OUTPUT statement to save the output as a data set and the PRINT procedure. For example, the following SAS statements print both the Error and Info columns for a power computation in a one-way ANOVA:

```sas
data MyExemp;
  input A $ Y1 Y2;
datalines;
  1 10 11
  2 12 11
  3 15 11
;
proc glmpower data=MyExemp;
  class A;
  model Y1 Y2 = A;
  power
    stddev = 2
    ntotal = 3 10
    power = .;
  ods output output=Power;
run;
proc print noobs data=Power;
  var NominalNTotal NTotal Dependent Power Error Info;
run;
```

The output is shown in Figure 50.5.

![Figure 50.5 Error and Information Columns](image-url)
The sample size of 3 specified with the NTOTAL= option causes an “Invalid input” message in the Error column and an “Error DF=0” message in the Info column, because a sample size of 3 is so small that there are no degrees of freedom left for the error term. The sample size of 10 causes an “Input N adjusted” message in the Info column, because it is rounded down to 9 to produce integer group sizes of 3 per cell. The cell means scenario represented by the dependent variable \( Y_2 \) causes a “No effect” message to appear in the Info column, because the means in this scenario are all equal.

Displayed Output

If you use the PLOTONLY option in the PROC GLMPOWER statement, the procedure displays only graphical output. Otherwise, the displayed output of the GLMPOWER procedure includes the following:

- the “Fixed Scenario Elements” table, which shows all applicable single-valued analysis parameters, in the following order: the dependent variable that represents the cell means scenario (for a univariate model) or the dependent variable transformation (for a multivariate model), the source of the test (that is, the model effect or between-subject contrast), the weight variable, parameters that are input explicitly, parameters that are supplied with defaults, and ancillary results

- an output table that shows the following when applicable (in order): the index of the scenario, the dependent variable that represents the cell means scenario (for a univariate model) or the dependent variable transformation (for a multivariate model), the type of the test, the source of the test (that is, the model effect or between-subject contrast), all multivalued input, ancillary results, the primary computed result, and error descriptions

- plots (if requested)

The exception to these ordering conventions is that the DEPENDENT and EFFECTS= options can be used along with the OUTPUTORDER=SYNTAX or OUTPUTORDER=REVERSE option in the POWER statement to specify the relative location of the output for dependent variable and type and source of test.

Ancillary results include the following:

- Actual Power, the achieved power, if it differs from the input (Nominal) power value
- Actual Alpha, the achieved significance level, if it differs from the input (Nominal) alpha value
- fractional sample size, if the NFRACTIONAL option is used in the POWER statement
- test or numerator degrees of freedom in the test’s critical value
- error or denominator degrees of freedom in the test’s critical value
- Effect, the combination of the within-subject Transformation contrast and between-subject Source test or contrast in a multivariate model

If sample size is the result parameter and the NFRACTIONAL option is used in the POWER statement, then both “Fractional” and “Ceiling” sample size results are displayed. Fractional sample sizes correspond to the “Nominal” values of power. Ceiling sample sizes are simply the fractional sample sizes rounded up to the nearest integer; they correspond to “Actual” values of power.
The noncentrality parameter is computed and stored in a hidden column called Noncentrality in the “Output” table. If a univariate test for a multivariate model is specified (that is, one of MTEST=BOX, MTEST=GG, MTEST=HF, or MTEST=UNCORR), then the numerator and denominator degrees of freedom that are used in the noncentral $F$ approximation of the test statistic distribution are computed and stored in hidden columns called NumNCDF and DenNCDF, respectively, in the “Output” table. These are the only tests for which the degrees of freedom in the noncentral $F$ approximation of the test statistic are different from those in the critical value.

**ODS Table Names**

PROC GLMPOWER assigns a name to each table that it creates. You can use these names to reference the table when using the Output Delivery System (ODS) to select tables and create output data sets. These names are listed in Table 50.7. For more information about ODS, see Chapter 20, “Using the Output Delivery System.”

<table>
<thead>
<tr>
<th>ODS Table Name</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>FixedElements</td>
<td>Factoid with single-valued analysis parameters</td>
</tr>
<tr>
<td>Output</td>
<td>All input and computed analysis parameters, error messages, and information messages for each scenario</td>
</tr>
<tr>
<td>PlotContent</td>
<td>Data contained in plots, including analysis parameters and indices identifying plot features. (NOTE: This table is saved as a data set and not displayed in PROC GLMPOWER output.)</td>
</tr>
</tbody>
</table>

**Computational Methods and Formulas**

This section describes the approaches that PROC GLMPOWER uses to compute power and sample size.

**Contrasts in Fixed-Effect Univariate Models**

The univariate linear model has the form

$$y = X\beta + \epsilon$$

where $y$ is the $N \times 1$ vector of responses, $X$ is the $N \times k$ design matrix, $\beta$ is the $k \times 1$ vector of model parameters that correspond to the columns of $X$, and $\epsilon$ is an $N \times 1$ vector of errors with

$$\epsilon_1, \ldots, \epsilon_N \sim N(0, \sigma^2) \quad \text{(iid)}$$

In PROC GLMPOWER, the model parameters $\beta$ are not specified directly, but rather indirectly as $y^*$, which represents either conjectured response means or typical response values for each design profile. The $y^*$
values are manifested as the dependent variable in the `MODEL` statement. The vector $\beta$ is obtained from $y^*$ according to the least squares equation,

$$\beta = (X'X)^{-1}X'y^*$$

Note that, in general, there is not a one-to-one mapping between $y^*$ and $\beta$. Many different scenarios for $y^*$ might lead to the same $\beta$. If you specify $y^*$ with the intention of representing cell means, keep in mind that PROC GLMPOWER allows scenarios that are not valid cell means according to the model that is specified in the `MODEL` statement. For example, if $y^*$ exhibits an interaction effect but the corresponding interaction term is left out of the model, then the cell means ($X\beta$) that are derived from $\beta$ differ from $y^*$. In particular, the cell means that are derived in this way are the projection of $y^*$ onto the model space.

It is convenient in power analysis to parameterize the design matrix $X$ in three parts, $\{\bar{X}, w, N\}$, defined as follows:

1. The $q \times k$ essence design matrix $\bar{X}$ is the collection of unique rows of $X$. Its rows are sometimes referred to as “design profiles.” Here, $q \leq N$ is defined simply as the number of unique rows of $X$.
2. The $q \times 1$ weight vector $w$ reveals the relative proportions of design profiles, and $W = \text{diag}(w)$. Row $i$ of $\bar{X}$ is to be included in the design $w_i$ times for every $w_j$ times that row $j$ is included. The weights are assumed to be standardized (that is, they sum up to 1).
3. The total sample size is $N$. This is the number of rows in $X$. If you gather $Nw_i = n_i$ copies of the $i$th row of $\bar{X}$, for $i = 1, \ldots, q$, then you end up with $X$.

The preceding quantities are derived from PROC GLMPOWER syntax as follows:

- Values for $\bar{X}$, $y^*$, and $w$ are specified in the exemplary data set (from using the `DATA=` option in the `PROC GLMPOWER` statement), and the corresponding variables are identified in the `CLASS`, `MODEL`, and `WEIGHT` statements.
- $N$ is specified in the `NTOTAL=` option in the `POWER` statement.

It is useful to express the crossproduct matrix $X'X$ in terms of these three parts,

$$X'X = N\bar{X}'W\bar{X}$$

because this expression factors out the portion $(N)$ that depends on sample size and the portion $(\bar{X}'W\bar{X})$ that depends only on the design structure.

A general linear hypothesis for the univariate model has the form

$$H_0: L\beta = \theta_0$$

$$H_A: L\beta \neq \theta_0$$

where $L$ is an $l \times k$ contrast matrix with rank $r_L$ and $\theta_0$ is the null value (usually just a vector of zeros).

Note that model effect tests are just contrasts that use special forms of $L$. Thus, this scheme covers both effect tests (which are specified in the `MODEL` statement and the `EFFECTS=` option in the `POWER` statement) and custom contrasts (which are specified in the `CONTRAST` statement).
The model degrees of freedom \( DF_M \) are equal to the rank of \( X \), denoted \( r_X \). The error degrees of freedom \( DF_E \) are equal to \( N - r_X \). The sample size \( N \) must be at least \( DF_M \) plus the number of covariates.

The test statistic is

\[
F = \frac{\left( SS_H \right)}{\hat{\sigma}^2}
\]

where

\[
SS_H = \frac{1}{N} \left( L\hat{\beta} - \theta_0 \right)^T \left( L (X'X)^{-1} L \right)^{-1} \left( L\hat{\beta} - \theta_0 \right)
\]

\[
\hat{\beta} = (X'X)^{-1} X'y
\]

\[
\hat{\sigma}^2 = \frac{1}{DF_E} \left( y - X\hat{\beta} \right)^T \left( y - X\hat{\beta} \right)
\]

Under \( H_0 \), \( F \sim F(r_L, DF_E) \). Under \( H_A \), \( F \) is distributed as \( F(r_L, DF_E, \lambda) \) with noncentrality

\[
\lambda = N (L\beta - \theta_0)^T \left( L \left( \hat{X}'W\hat{X} \right)^{-1} L \right)^{-1} (L\beta - \theta_0) \sigma^{-2}
\]

The value of \( \sigma \) is specified in the STDDEV= option in the POWER statement.

Muller and Peterson (1984) give the exact power of the test as

\[
\text{power} = P \left( F(r_L, DF_E, \lambda) \geq F_{1-\alpha}(r_L, DF_E) \right)
\]

The value of \( \alpha \) is specified in the ALPHA= option in the POWER statement.

Sample size is computed by inverting the power equation.

See Muller and Benignus (1992) and O’Brien and Shieh (1992) for additional discussion.

### Adjustments for Covariates in Univariate Models

If you specify covariates in a univariate model (whether continuous or categorical), then two adjustments are made in order to compute approximate power in the presence of the covariates. Let \( n_v \) denote the number of covariates (counting dummy variables for categorical covariates individually) as specified in the NCOVARIATES= option in the POWER statement. In other words, \( n_v \) is the total degrees of freedom used by the covariates. The adjustments are as follows:

1. The error degrees of freedom decrease by \( n_v \).

2. The error standard deviation \( \sigma \) shrinks by a factor of \( (1 - \rho^2)^{\frac{1}{2}} \) (if the CORRXY= option is used to specify the correlation \( \rho \) between covariates and response) or \( (1 - r)^{\frac{1}{2}} \) (if the PROPVARREDUCTION= option is used to specify the proportional reduction in total \( R^2 \) incurred by the covariates). Let \( \sigma^* \) represent the updated value of \( \sigma \).

As a result of these changes, the power is computed as

\[
\text{power} = P \left( F(r_L, DF_E - n_v, \lambda^*) \geq F_{1-\alpha}(r_L, N - r_X - n_v) \right)
\]

where \( \lambda^* \) is calculated using \( \sigma^* \) rather than \( \sigma \):

\[
\lambda^* = N (L\beta - \theta_0)^T \left( L \left( \hat{X}'W\hat{X} \right)^{-1} L \right)^{-1} (L\beta - \theta_0) (\sigma^*)^{-2}
\]
Contrasts in Fixed-Effect Multivariate Models

The multivariate model has the form

\[ Y = X\beta + \epsilon \]

where \( Y \) is the \( N \times p \) vector of responses, for \( p > 1 \); \( X \) is the \( N \times k \) design matrix; \( \beta \) is the \( k \times p \) matrix of model parameters that correspond to the columns of \( X \) and \( Y \); and \( \epsilon \) is an \( N \times p \) vector of errors, where

\[ \epsilon_1, \ldots, \epsilon_N \sim N(0, \Sigma) \quad (\text{iid}) \]

In PROC GLMPOWER, the model parameters \( \beta \) are not specified directly, but rather indirectly as \( Y^* \), which represents either conjectured response means or typical response values for each design profile. The \( Y^* \) values are manifested as the collection of dependent variables in the MODEL statement. The matrix \( \beta \) is obtained from \( Y^* \) according to the least squares equation,

\[ \beta = (X'X)^{-1}X'Y^* \]

Note that, in general, there is not a one-to-one mapping between \( Y^* \) and \( \beta \). Many different scenarios for \( Y^* \) might lead to the same \( \beta \). If you specify \( Y^* \) with the intention of representing cell means, keep in mind that PROC GLMPOWER allows scenarios that are not valid cell means according to the model that is specified in the MODEL statement. For example, if \( Y^* \) exhibits an interaction effect but the corresponding interaction term is left out of the model, then the cell means \( (X\beta) \) that are derived from \( \beta \) differ from \( Y^* \). In particular, the cell means that are derived in this way are the projection of \( Y^* \) onto the model space.

It is convenient in power analysis to parameterize the design matrix \( X \) in three parts, \( \{\tilde{X}, W, N\} \), defined as follows:

1. The \( q \times k \) essence design matrix \( \tilde{X} \) is the collection of unique rows of \( X \). Its rows are sometimes referred to as “design profiles.” Here, \( q \leq N \) is defined simply as the number of unique rows of \( X \).

2. The \( q \times 1 \) weight vector \( w \) reveals the relative proportions of design profiles, and \( W = \text{diag}(w) \). Row \( i \) of \( \tilde{X} \) is to be included in the design \( w_i \) times for every \( w_j \) times that row \( j \) is included. The weights are assumed to be standardized (that is, they sum up to 1).

3. The total sample size is \( N \). This is the number of rows in \( X \). If you gather \( Nw_i = n_i \) copies of the \( i \)th row of \( \tilde{X} \), for \( i = 1, \ldots, q \), then you end up with \( X \).

The preceding quantities are derived from PROC GLMPOWER syntax as follows:

- Values for \( \tilde{X}, Y^* \), and \( w \) are specified in the exemplary data set (from using the DATA= option in the PROC GLMPOWER statement), and the corresponding variables are identified in the CLASS, MODEL, and WEIGHT statements.

- \( N \) is specified in the NTOTAL= option in the POWER statement.

It is useful to express the crossproduct matrix \( X'X \) in terms of these three parts,

\[ X'X = N\tilde{X}'W\tilde{X} \]

because this expression factors out the portion \((N)\) that depends on sample size and the portion \((\tilde{X}'W\tilde{X})\) that depends only on the design structure.
A general linear hypothesis for the univariate model has the form

\[ H_0 : L \beta M = \theta_0 \]
\[ H_A : L \beta M \neq \theta_0 \]

where \( L \) is an \( l \times k \) between-subject contrast matrix with rank \( r_L \), \( M \) is a \( p \times m \) within-subject contrast matrix with rank \( r_M \), and \( \theta_0 \) is an \( l \times m \) null contrast matrix (usually just a matrix of zeros).

Note that model effect tests are just between-subject contrasts that use special forms of \( L \), combined with an \( M \) that is the \( p \times 1 \) mean transformation vector of the dependent variables (a vector of values all equal to \( 1/p \)). Thus, this scheme covers both effect tests (which are specified in the MODEL statement and the EFFECTS= option in the POWER statement) and custom between-subject contrasts (which are specified in the CONTRAST statement).

The \( M \) matrix is often referred to as the dependent variable transformation and is specified in the MANOVA or REPEATED statement.

The model degrees of freedom \( DF_M \) are equal to the rank of \( X \), denoted \( r_X \). The error degrees of freedom \( DF_E \) are equal to \( N - r_X \).

The hypothesis sum of squares \( SS_H \) in the univariate model generalizes to the hypothesis SSCP matrix in the multivariate model,

\[ H = (L \hat{\beta} M - \theta_0)' (L (X'X)^{-1} L')^{-1} (L \hat{\beta} M - \theta_0) \]

The error sum of squares \( \hat{\sigma}^2 (N - r_X) \) in the univariate model generalizes to the error SSCP matrix in the multivariate model,

\[ E = (N - r_X) M' \hat{\Sigma} M \]

where

\[ \hat{\Sigma} = (Y - X\hat{\beta})' (Y - X\hat{\beta}) /(N - r_X) \]

and

\[ \hat{\beta} = (X'X)^{-1} X'Y \]

The population counterpart of \( H/N \) is

\[ H^* = (L \beta M - \theta_0)' (L (\bar{X}' W \bar{X})^{-1} L')^{-1} (L \beta M - \theta_0) \]

and the population counterpart of \( E/N \) is

\[ E^* = M' \Sigma M \]

The elements of \( \Sigma \) are specified in the MATRIX= and STDDEV= options and identified in the CORRMAT=, CORRS=, COVMAT=, and SQRTVAR= options in the POWER statement.

The power and sample size computations for all the tests that are supported in the MTEST= option in the POWER statement are based on \( H^* \) and \( E^* \). The following two subsections cover the computational methods and formulas for the multivariate and univariate tests that are supported in the MTEST= and UEPSDEF= options in the POWER statement.
### Multivariate Tests

Power computations for multivariate tests are based on O’Brien and Shieh (1992) (for METHOD=OBRIENSHIEH) and Muller and Peterson (1984) (for METHOD=MULLERPETERSON).

Let \( s = \min(r_L, r_M) \), the smaller of the between-subject and within-subject contrast degrees of freedom. Critical value computations assume that under \( H_0 \), the test statistic \( F \) is distributed as \( F(r_L r_M, v_2) \), where \( v_2 = (N - r_X) - r_M + 1 \) if \( s = 1 \) but depends on the choice of test if \( s > 1 \). Power computations assume that under \( H_A \), \( F \) is distributed as \( F(r_L r_M, v_2, \lambda) \), where the noncentrality \( \lambda \) depends on \( r_L, r_M \), the choice of test, and the power computation method.

Formulas for the test statistic \( F \), denominator degrees of freedom \( v_2 \), and noncentrality \( \lambda \) for all combinations of dimensions, tests, and methods are given in the following subsections.

The power in each case is computed as

\[
\text{power} = P \left( F(r_L r_M, v_2, \lambda) \geq F_{1-\alpha}(r_L r_M, v_2) \right)
\]

Computed power is exact for some cases and approximate for others. Sample size is computed by inverting the power equation.

Let \( \Delta = E^{-1}H \), and define \( \phi \) as the \( s \times 1 \) vector of ordered positive eigenvalues of \( \Delta \), \( \phi = \{\phi_1, \ldots, \phi_s\} \), where \( \phi_1 \geq \cdots \geq \phi_s > 0 \). The population equivalent is

\[
\Delta^* = E^{*-1}H^* \\
= (M^T \Sigma M)^{-1} (L \beta M - \theta_0)^T \left( L^T W L \right)^{-1} L^T (L \beta M - \theta_0)
\]

where \( \phi^* \) is the \( s \times 1 \) vector of ordered positive eigenvalues of \( \Delta^* \), \( \phi^* = \{\phi_1^*, \ldots, \phi_s^*\} \) for \( \phi_1^* \geq \cdots \geq \phi_s^* > 0 \).

#### Case 1: \( s = 1 \)

When \( s = 1 \), all three multivariate tests (MTEST=HLT, MTEST=PT, and MTEST=WILKS) are equivalent. The test statistic is \( F = \phi_1 v_2 / (r_L r_M) \), where \( v_2 = (N - r_X) - r_M + 1 \).

When the dependent variable transformation has a single degree of freedom (\( r_M = 1 \), METHOD=OBRIENSHIEH and METHOD=MULLERPETERSON are the same, computing exact power by using noncentrality \( \lambda = N \Delta^* \). The sample size must satisfy \( N \geq r_X + 1 \).

When the dependent variable transformation has more than one degree of freedom but the between-subject contrast has a single degree of freedom (\( r_M > 1, r_L = 1 \), METHOD=OBRIENSHIEH computes exact power by using noncentrality \( \lambda = N \phi_1^* \), and METHOD=MULLERPETERSON computes approximate power by using

\[
\lambda = \frac{(N - r_X) - r_M + 1}{(N - r_X)} N \phi_1^*
\]

The sample size must satisfy \( N \geq r_X + r_M \).

#### Case 2: \( s > 1 \)

When both the dependent variable transformation and the between-subject contrast have more than one degree of freedom (\( s > 1 \), METHOD=OBRIENSHIEH computes the noncentrality as \( \lambda = N \lambda^* \), where \( \lambda^* \) is the primary noncentrality. The form of \( \lambda^* \) depends on the choice of test statistic.
METHOD=MULLERPETERSON computes the noncentrality as \( \lambda = v_2 \lambda^{(MP)}^* \), where \( \lambda^{(MP)}^* \) has the same form as \( \lambda^* \) except that \( \phi^* \) is replaced by

\[
\phi^{(MP)}^* = \frac{N}{(N - r_X)} \phi^*
\]

Computed power is approximate for both methods when \( s > 1 \).

**Hotelling-Lawley Trace (MTEST=HLT) When \( s > 1 \)**

If \( N > r_X + r_M + 1 \), then the denominator degrees of freedom for the Hotelling-Lawley trace are \( v_2 = v_{2a} \).

\[
v_{2a} = 4 + (r_L r_M + 2)g
\]

where

\[
g = \frac{(N - r_X)^2 - (N - r_X)(2r_M + 3) + r_M (r_M + 3)}{(N - r_X)(r_L + r_M + 1) - (r_L + 2r_M + r_M^2 - 1)}
\]

which is the same as \( v_2^{(T_2)} \) in O’Brien and Shieh (1992) and is due to McKeon (1974).

If \( N \leq r_X + r_M + 1 \), then \( v_2 = v_{2b} \).

\[
v_{2b} = s((N - r_X) - r_M - 1) + 2
\]

which is the same as both \( v_2^{(T_1)} \) in O’Brien and Shieh (1992) and \( v_2 \) in Muller and Peterson (1984) and is due to Pillai and Samson (1959).

The primary noncentrality is

\[
\lambda^* = \sum_{i=1}^{s} \phi_i^*
\]

The sample size must satisfy

\[
N \geq r_X + r_M + 1 - 1/s
\]

If \( N > r_X + r_M + 1 \), then the test statistic is

\[
F = \frac{U / v_1}{c / v_{2a}}
\]

where

\[
U = \text{trace}(E^{-1}H)
\]

\[
= \sum_{i=1}^{s} \phi_i
\]

and

\[
c = \frac{2 + (r_L r_M + 2)g}{N - r_X - r_M - 1}
\]
If \( N \leq r_X + r_M + 1 \), then the test statistic is
\[
F = \frac{U/v_1}{s/v_{2b}}
\]

**Pillai’s Trace (MTEST=PT) When \( s > 1 \)**

The denominator degrees of freedom for Pillai’s trace are
\[
\nu_2 = s((N - r_X) + s - r_M)
\]
The primary noncentrality is
\[
\lambda^* = s \left( \frac{\sum_{i=1}^{s} \frac{\phi_i^*}{1 + \phi_i^*}}{s - \sum_{i=1}^{s} \frac{\phi_i^*}{1 + \phi_i^*}} \right)
\]
The sample size must satisfy
\[
N \geq r_X + r_M + 1/s - s
\]
The test statistic is
\[
F = \frac{V/v_1}{(s - V)/\nu_2}
\]
where
\[
V = \text{trace} (H(H + E)^{-1})
\]
\[
= \sum_{i=1}^{s} \frac{\phi_i}{1 + \phi_i}
\]

**Wilks’ Lambda (MTEST=WILKS) When \( s > 1 \)**

The denominator degrees of freedom for Wilks’ lambda are
\[
\nu_2 = t \left[ (N - r_X) - 0.5(r_M - r_L + 1) - 0.5(r_L r_M - 2) \right]
\]
where
\[
t = \begin{cases} 
1 & \text{if } r_L r_M \leq 3 \\
\left[ \frac{(r_L r_M)^2 - 4}{r_L^2 + 2r_M^2 - 5} \right]^{1/2} & \text{if } r_L r_M \geq 4
\end{cases}
\]
The primary noncentrality is
\[
\lambda^* = t \left[ \left( \prod_{i=1}^{s} \left( (1 + \phi_i^*)^{-1} \right)^{-\frac{1}{t}} \right) - 1 \right]
\]
The sample size must satisfy
\[
N \geq (1 + 0.5(r_L r_M - 2))/t + r_X + (r_M - r_L + 1)/2
\]
The test statistic is
\[ F = \frac{(1 - \Lambda^{1/t})/\nu_1}{\Lambda^{1/t}/\nu_2} \]
where
\[ \Lambda = \text{det}(E)/\text{det}(H + E) \]
\[ = \prod_{i=1}^{s} ((1 + \phi_i)^{-1}) \]

**Univariate Tests**

Power computations for univariate tests are based on Muller et al. (2007) and Muller and Barton (1989).

The test statistic is
\[ F = \frac{\text{trace}(H)/r_L}{\text{trace}(E)/(N - r_X)} \]

Critical value computations assume that under \( H_0 \), \( F \) is distributed as \( F(\nu_1, \nu_2) \), where \( \nu_1 \) and \( \nu_2 \) depend on the choice of test.

The four tests for the univariate approach to repeated measures differ in their assumptions about the sphericity \( \varepsilon \) of \( E^* \),
\[ \varepsilon = \frac{\text{trace}^2(E^*)}{r_M \text{trace}(E^{*2})} \]

Power computations assume that under \( H_A \), \( F \) is distributed as \( F(\nu_1^*, \nu_2^*, \lambda) \).

Formulas for \( \nu_1 \) and \( \nu_2 \) for each test and formulas for \( \nu_1^*, \nu_2^* \), and \( \lambda \) are given in the following subsections.

The power in each case is approximated as
\[ \text{power} = P\left( F(\nu_1^*, \nu_2^*, \lambda) \geq F_{1-\alpha}(\nu_1, \nu_2) \right) \]

Sample size is computed by inverting the power equation.

The sample size must be large enough to yield \( \nu_1 > 0, \nu_1^* > 0, \nu_2 \geq 1, \) and \( \nu_2^* \geq 1 \).

Because these univariate tests are biased, the achieved significance level might differ from the nominal significance level. The actual alpha is computed in the same way as the power, except that the noncentrality parameter \( \lambda \) is set to 0.

Define \( \phi^{(E)} \) as the vector of ordered eigenvalues of \( E^* \), \( \phi^{(E)} = \{\phi_1^{(E)}, \ldots, \phi_{r_M}^{(E)}\} \), where \( \phi_1^{(E)} \geq \cdots \geq \phi_{r_M}^{(E)} \), and define \( \gamma_j^{(E)} \) as the \( j \)th eigenvector of \( E^* \). Critical values and power computations are based on the following intermediate parameters:
\[ \omega_{*j} = N \left( \gamma_j^{(E)} \right)' \mathbf{H}^* \gamma_j^{(E)}/\phi_j^{(E)} \]
The degrees of freedom and noncentrality in the noncentral F approximation of the test statistic are computed as follows:

\[ S_{t1} = \sum_{j=1}^{r_M} \phi_j^{(E)} \]
\[ S_{t2} = \sum_{j=1}^{r_M} \phi_j^{(E)} \omega_{*j} \]
\[ S_{t3} = \sum_{j=1}^{r_M} \left( \phi_j^{(E)} \right)^2 \]
\[ S_{t4} = \sum_{j=1}^{r_M} \left( \phi_j^{(E)} \right)^2 \omega_{*j} \]

\[ R_{*1} = \frac{r_L S_{t3} + 2S_{t4}}{r_L S_{t1} + 2S_{t2}} \]
\[ R_{*2} = \frac{S_{t3}}{S_{t1}} \]

\[ E(t_1) = 2(N - r_X)S_{t3} + (N - r_X)^2 S_{t1}^2 \]
\[ E(t_2) = (N - r_X)((N - r_X) + 2)S_{t3} + 2(N - r_X) \sum_{j_1=2}^{r_M} \sum_{j_2=1}^{j_1-1} \phi_{j_1}^{(E)} \phi_{j_2}^{(E)} \]

The uncorrected test assumes sphericity \( \varepsilon = 1 \), in which case the null F distribution is exact, with the following degrees of freedom:

\[ \nu_1^* = \frac{r_L S_{t1}}{R_{*1}} \]
\[ \nu_2^* = \frac{(N - r_X)S_{t1}}{R_{*2}} \]
\[ \lambda = \frac{S_{t2}}{R_{*1}} \]

**Uncorrected Test**

The uncorrected test assumes sphericity \( \varepsilon = 1 \), in which case the null F distribution is exact, with the following degrees of freedom:

\[ \nu_1 = r_L r_M \]
\[ \nu_2 = r_M (N - r_X) \]

**Greenhouse-Geisser Adjustment (MTEST=UNCORR)**

The Greenhouse-Geisser adjustment to the uncorrected test reduces degrees of freedom by the MLE \( \hat{\varepsilon} \) of the sphericity,

\[ \hat{\varepsilon} = \frac{\text{trace}^2(E)}{r_M \text{trace}(E^2)} \]
An approximation for the expected value of $\hat{\epsilon}$ is used to compute the degrees of freedom for the null $F$ distribution,

$$v_1 = r_L r_M E(\hat{\epsilon})$$
$$v_2 = r_M (N - r_X) E(\hat{\epsilon})$$

where

$$E(\hat{\epsilon}) = \frac{E(t_1)}{r_M E(t_2)}$$

**Huynh-Feldt Adjustments (MTEST=HF)**

The Huynh-Feldt adjustment reduces degrees of freedom by a nearly unbiased estimate $\tilde{\epsilon}$ of the sphericity,

$$\tilde{\epsilon} = \begin{cases} 
\frac{N r_M \hat{\epsilon} - 2}{r_M ((N - r_X) - r_M \hat{\epsilon})} & \text{if UEPSDEF=HF} \\
\frac{r_M ((N - r_X) - r_M \hat{\epsilon} = 2)}{r_M ((N - r_X) - r_M \hat{\epsilon} = 2)} & \text{if UEPSDEF=HFL} \\
\left(\frac{(v_a - 2)(v_a - 4)}{v_a^2}\right) & \text{if UEPSDEF=CM}
\end{cases}$$

where

$$v_a = (N - r_X - 1) + (N - r_X)(N - r_X - 1)/2$$

The value of $\tilde{\epsilon}$ is truncated if necessary to be at least $1/r_M$ and at most 1.

An approximation for the expected value of $\tilde{\epsilon}$ is used to compute the degrees of freedom for the null $F$ distribution,

$$v_1 = r_L r_M E_t(\tilde{\epsilon})$$
$$v_2 = r_M (N - r_X) E_t(\tilde{\epsilon})$$

where

$$E_t(\tilde{\epsilon}) = \min(\max(E(\hat{\epsilon}), 1/r_M), 1)$$

and

$$E(\tilde{\epsilon}) = \begin{cases} 
\frac{N E(t_1) - 2E(t_2)}{r_M ((N - r_X) E(t_2) - E(t_1))} & \text{if UEPSDEF=HF} \\
\frac{r_M ((N - r_X) E(t_2) - E(t_1))}{r_M ((N - r_X) E(t_2) - E(t_1))} & \text{if UEPSDEF=HFL} \\
\left(\frac{(v_a - 2)(v_a - 4)}{v_a^2}\right) & \text{if UEPSDEF=CM}
\end{cases}$$

**Box Conservative Test (MTEST=BOX)**

The Box conservative test assumes the worst case for sphericity, $\varepsilon = 1/r_M$, leading to the following degrees of freedom for the null $F$ distribution:

$$v_1 = r_L$$
$$v_2 = (N - r_X)$$
ODS Graphics

Statistical procedures use ODS Graphics to create graphs as part of their output. ODS Graphics is described in detail in Chapter 21, “Statistical Graphics Using ODS.”

Before you create graphs, ODS Graphics must be enabled (for example, by specifying the ODS GRAPHICS ON statement). For more information about enabling and disabling ODS Graphics, see the section “Enabling and Disabling ODS Graphics” on page 615 in Chapter 21, “Statistical Graphics Using ODS.”

The overall appearance of graphs is controlled by ODS styles. Styles and other aspects of using ODS Graphics are discussed in the section “A Primer on ODS Statistical Graphics” on page 614 in Chapter 21, “Statistical Graphics Using ODS.”

If ODS Graphics is not enabled, then PROC GLMPOWER creates traditional graphics.

You can reference every graph produced through ODS Graphics with a name. The names of the graphs that PROC GLMPOWER generates are listed in Table 50.8, along with the required statements and options.

<table>
<thead>
<tr>
<th>ODS Graph Name</th>
<th>Plot Description</th>
<th>Option</th>
</tr>
</thead>
<tbody>
<tr>
<td>PowerPlot</td>
<td>Plot with power and sample size on the axes</td>
<td>PLOT</td>
</tr>
<tr>
<td>PowerAbort</td>
<td>Empty plot that shows an error message when a plot could not be produced</td>
<td>PLOT</td>
</tr>
</tbody>
</table>

Examples: GLMPOWER Procedure

Example 50.1: One-Way ANOVA

This example deals with the same situation as in Example 91.1 in Chapter 91, “The POWER Procedure.”

Hocking (1985, p. 109) describes a study of the effectiveness of electrolytes in reducing lactic acid buildup for long-distance runners. You are planning a similar study in which you will allocate five different fluids to runners on a 10-mile course and measure lactic acid buildup immediately after the race. The fluids consist of water and two commercial electrolyte drinks, EZDure and LactoZap, each prepared at two concentrations, low (EZD1 and LZ1) and high (EZD2 and LZ2).

You conjecture that the standard deviation of lactic acid measurements given any particular fluid is about 3.75, and that the expected lactic acid values will correspond roughly to Table 50.9. You are least familiar with the LZ1 drink and hence decide to consider a range of reasonable values for that mean.
Table 50.9  Mean Lactic Acid Buildup by Fluid

<table>
<thead>
<tr>
<th></th>
<th>Water</th>
<th>EZD1</th>
<th>EZD2</th>
<th>LZ1</th>
<th>LZ2</th>
</tr>
</thead>
<tbody>
<tr>
<td>Value</td>
<td>35.6</td>
<td>33.7</td>
<td>30.2</td>
<td>29 or 28</td>
<td>25.9</td>
</tr>
</tbody>
</table>

You are interested in four different comparisons, shown in Table 50.10 with appropriate contrast coefficients.

Table 50.10  Planned Comparisons

<table>
<thead>
<tr>
<th>Comparison</th>
<th>Water</th>
<th>EZD1</th>
<th>EZD2</th>
<th>LZ1</th>
<th>LZ2</th>
</tr>
</thead>
<tbody>
<tr>
<td>Water versus electrolytes</td>
<td>4</td>
<td>-1</td>
<td>-1</td>
<td>-1</td>
<td>-1</td>
</tr>
<tr>
<td>EZD versus LZ</td>
<td>0</td>
<td>1</td>
<td>1</td>
<td>-1</td>
<td>-1</td>
</tr>
<tr>
<td>EZD1 versus EZD2</td>
<td>0</td>
<td>1</td>
<td>-1</td>
<td>0</td>
<td>0</td>
</tr>
<tr>
<td>LZ1 versus LZ2</td>
<td>0</td>
<td>0</td>
<td>0</td>
<td>1</td>
<td>-1</td>
</tr>
</tbody>
</table>

For each of these contrasts you want to determine the sample size required to achieve a power of 0.9 for detecting an effect with magnitude in accord with Table 50.9. You are not yet attempting to choose a single sample size for the study, but rather checking the range of sample sizes needed for individual contrasts. You plan to test each contrast at $\alpha = 0.025$. In the interests of reducing costs, you will provide twice as many runners with water as with any of the electrolytes; that is, you will use a sample size weighting scheme of 2:1:1:1:1.

Before calling PROC GLMPOWER, you need to create the *exemplary data set* to specify means and weights for the design profiles:

```sas
data Fluids;
  input Fluid $ LacticAcid1 LacticAcid2 CellWgt;
  datalines;
  Water 35.6 35.6 2
  EZD1 33.7 33.7 1
  EZD2 30.2 30.2 1
  LZ1 29 28 1
  LZ2 25.9 25.9 1
;
```

The variable LacticAcid1 represents the cell means scenario with the larger LZ1 mean (29), and LacticAcid2 represents the scenario with the smaller LZ1 mean (28). The variable CellWgt contains the sample size allocation weights.

Use the DATA= option in the PROC GLMPOWER statement to specify Fluids as the exemplary data set. The following statements perform the sample size analysis:

```sas
proc glmpower data=Fluids;
  class Fluid;
  model LacticAcid1 LacticAcid2 = Fluid;
  weight CellWgt;
  contrast "Water vs. others" Fluid -1 -1 -1 -1 4;
  contrast "EZD vs. LZ" Fluid 1 1 -1 -1 0;
```
Example 50.1: One-Way ANOVA

```plaintext
contrast "EZD1 vs. EZD2" Fluid 1 -1 0 0 0;
contrast "LZ1 vs. LZ2" Fluid 0 0 1 -1 0;
power
   stddev = 3.75
   alpha  = 0.025
   ntotal = .
   power  = 0.9;
run;
```

The `CLASS` statement identifies `Fluid` as a classification variable. The `MODEL` statement specifies the model and the two cell means scenarios `LacticAcid1` and `LacticAcid2`. The `WEIGHT` statement identifies `CellWgt` as the weight variable. The `CONTRAST` statement specifies the contrasts. Since `PROC GLMPOWER` by default processes class levels in order of formatted values, the contrast coefficients correspond to the following order: EZD1, EZD2, LZ1, LZ2, Water. (Note: You could use the `ORDER=DATA` option in the `PROC GLMPOWER` statement to achieve the same ordering as in Table 50.10 instead.) The `POWER` statement specifies total sample size as the result parameter and provides values for the other analysis parameters (error standard deviation, alpha, and power).

Output 50.1.1 displays the results.

Output 50.1.1  Sample Sizes for One-Way ANOVA Contrasts

The **GLMPOWER** Procedure

<table>
<thead>
<tr>
<th>Fixed Scenario Elements</th>
</tr>
</thead>
<tbody>
<tr>
<td>Weight Variable</td>
</tr>
<tr>
<td>Alpha</td>
</tr>
<tr>
<td>Error Standard Deviation</td>
</tr>
<tr>
<td>Nominal Power</td>
</tr>
</tbody>
</table>

<p>| Computed N Total |
|------------------|-----------------|-----------------|-----------------|</p>
<table>
<thead>
<tr>
<th>Index</th>
<th>Dependent</th>
<th>Type</th>
<th>Source</th>
<th>Test DF</th>
<th>Error DF</th>
<th>Actual Power</th>
<th>N Total</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>LacticAcid1</td>
<td>Effect</td>
<td>Fluid</td>
<td>4</td>
<td>25</td>
<td>0.958</td>
<td>30</td>
</tr>
<tr>
<td>2</td>
<td>LacticAcid1</td>
<td>Contrast</td>
<td>Water vs. others</td>
<td>1</td>
<td>25</td>
<td>0.947</td>
<td>30</td>
</tr>
<tr>
<td>3</td>
<td>LacticAcid1</td>
<td>Contrast</td>
<td>EZD vs. LZ</td>
<td>1</td>
<td>55</td>
<td>0.929</td>
<td>60</td>
</tr>
<tr>
<td>4</td>
<td>LacticAcid1</td>
<td>Contrast</td>
<td>EZD1 vs. EZD2</td>
<td>1</td>
<td>169</td>
<td>0.901</td>
<td>174</td>
</tr>
<tr>
<td>5</td>
<td>LacticAcid1</td>
<td>Contrast</td>
<td>LZ1 vs. LZ2</td>
<td>1</td>
<td>217</td>
<td>0.902</td>
<td>222</td>
</tr>
<tr>
<td>6</td>
<td>LacticAcid2</td>
<td>Effect</td>
<td>Fluid</td>
<td>4</td>
<td>25</td>
<td>0.972</td>
<td>30</td>
</tr>
<tr>
<td>7</td>
<td>LacticAcid2</td>
<td>Contrast</td>
<td>Water vs. others</td>
<td>1</td>
<td>19</td>
<td>0.901</td>
<td>24</td>
</tr>
<tr>
<td>8</td>
<td>LacticAcid2</td>
<td>Contrast</td>
<td>EZD vs. LZ</td>
<td>1</td>
<td>43</td>
<td>0.922</td>
<td>48</td>
</tr>
<tr>
<td>9</td>
<td>LacticAcid2</td>
<td>Contrast</td>
<td>EZD1 vs. EZD2</td>
<td>1</td>
<td>169</td>
<td>0.901</td>
<td>174</td>
</tr>
<tr>
<td>10</td>
<td>LacticAcid2</td>
<td>Contrast</td>
<td>LZ1 vs. LZ2</td>
<td>1</td>
<td>475</td>
<td>0.902</td>
<td>480</td>
</tr>
</tbody>
</table>

The sample sizes range from 24 for the comparison of water versus electrolytes to 480 for the comparison of LZ1 versus LZ2, both assuming the smaller LZ1 mean. The sample size for the latter comparison is relatively large because the small mean difference of 28 - 25.9 = 2.1 is hard to detect. `PROC GLMPOWER` also includes the effect test for `Fluid`. Note that, in this case, it is equivalent to `TEST=OVERALL_F` in the `ONEWAYANOV A` statement of `PROC POWER`, since there is only one effect in the model.

The Nominal Power of 0.9 in the “Fixed Scenario Elements” table in Output 50.1.1 represents the input target power, and the Actual Power column in the “Computed N Total” table is the power at the sample size. 
(N Total) adjusted to achieve the specified sample weighting. Note that all of the sample sizes are rounded up to multiples of 6 to preserve integer group sizes (since the group weights add up to 6). You can use the NFRACTIONAL option in the POWER statement to compute raw fractional sample sizes.

Suppose you want to plot the required sample size for the range of power values from 0.5 to 0.95. First, define the analysis by specifying the same statements as before, but add the PLOTONLY option to the PROC GLMPOWER statement to disable the nongraphical results. Next, specify the PLOT statement with X=POWER to request a plot with power on the X axis. (The result parameter—here sample size—is always plotted on the other axis.) Use the MIN= and MAX= options in the PLOT statement to specify the power range. The following statements produce the plot:

```plaintext
ods graphics on;
proc glmpower data=Fluids plotonly;
  class Fluid;
  model LacticAcid1 LacticAcid2 = Fluid;
  weight CellWgt;
  contrast "Water vs. others" Fluid -1 -1 -1 -1 4;
  contrast "EZD vs. LZ" Fluid 1 1 -1 -1 0;
  contrast "EZD1 vs. EZD2" Fluid 1 -1 0 0 0;
  contrast "LZ1 vs. LZ2" Fluid 0 0 1 -1 0;
  power
    stddev = 3.75
    alpha = 0.025
    ntotal = .
    power = 0.9;
  plot x=power min=.5 max=.95;
run;
```
Example 50.1: One-Way ANOVA

Output 50.1.2  Plot of Sample Size versus Power for One-Way ANOVA Contrasts

In Output 50.1.2, the line style identifies the cell means scenario, and the plotting symbol identifies the test. The plotting symbol locations identify actual computed powers; the curves are linear interpolations of these points. The plot shows that the required sample size is highest for the test of LZ1 versus LZ2, which was previously found to require the most resources.

Note that some of the plotted points in Output 50.1.2 are unevenly spaced. This is because the plotted points are the rounded sample size results at their corresponding actual power levels. The range specified with the MIN= and MAX= values in the PLOT statement corresponds to nominal power levels. In some cases, actual power is substantially higher than nominal power. To obtain plots with evenly spaced points (but with fractional sample sizes at the computed points), you can use the NFRACIONAL option in the POWER statement preceding the PLOT statement.

Finally, suppose you want to plot the power for the range of sample sizes you will likely consider for the study (the range of 24 to 480 that achieves 0.9 power for different comparisons). In the POWER statement, identify power as the result (POWER=.), and specify any total sample size value (say, NTOTAL=100). Specify the PLOT statement with X=N to request a plot with sample size on the X axis.

The following statements produce the plot:
proc glmpower data=Fluids plotonly;
class Fluid;
model LacticAcid1 LacticAcid2 = Fluid;
weight CellWgt;
contrast "Water vs. others" Fluid -1 -1 -1 -1 4;
contrast "EZD vs. LZ" Fluid 1 1 -1 -1 0;
contrast "EZD1 vs. EZD2" Fluid 1 -1 0 0 0;
contrast "LZ1 vs. LZ2" Fluid 0 0 1 -1 0;
power
  stddev = 3.75
  alpha  = 0.025
  ntotal = 24
  power  = .;
  plot x=n min=24 max=480;
run;
ods graphics off;

Note that the value 100 specified with the NTOTAL=100 option is not used. It is overridden in the plot by the MIN= and MAX= options in the PLOT statement, and the PLOTONLY option in the PROC GLMPOWER statement disables nongraphical results. But the NTOTAL= option (along with a value) is still needed in the POWER statement as a placeholder, to identify the desired parameterization for sample size.

See Output 50.1.3 for the plot.
Example 50.2: Two-Way ANOVA with Covariate

Suppose you can enhance the planned study discussed in Example 50.1 in two ways:

- incorporate results from races at two different altitudes (“high” and “low”)
- measure the body mass index of each runner before the race

This is equivalent to adding a second fixed effect and a continuous covariate to your model.

Since lactic acid buildup is more pronounced at higher altitudes, you will include altitude as a factor in the model along with fluid, extending the one-way ANOVA to a two-way ANOVA. In doing so, you expect to
lower the residual standard deviation from about 3.75 to 3.5 (in addition to generalizing the study results). You assume there is negligible interaction between fluid and altitude and plan to use a main-effects-only model. You conjecture that the mean lactic acid buildup follows Table 50.11.

### Table 50.11 Mean Lactic Acid Buildup by Fluid and Altitude

<table>
<thead>
<tr>
<th>Fluid</th>
<th>Altitude</th>
<th>Water</th>
<th>EZD1</th>
<th>EZD2</th>
<th>LZ1</th>
<th>LZ2</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>High</td>
<td>36.9</td>
<td>35.0</td>
<td>31.5</td>
<td>30</td>
<td>27.1</td>
</tr>
<tr>
<td></td>
<td>Low</td>
<td>34.3</td>
<td>32.4</td>
<td>28.9</td>
<td>27</td>
<td>24.7</td>
</tr>
</tbody>
</table>

By including a measurement of body mass index as a covariate in the study, you hope to further reduce the error variability. The extent of this reduction in variability is commonly expressed in two alternative ways: (1) the correlation between the covariates and the response or (2) the proportional reduction in total R square incurred by the covariates. You prefer the former and guess that the correlation between body mass index and lactic acid buildup is between 0.2 and 0.3. You specify these estimates with the NCOVARIATES= and CORRXY= options in the POWER statement. The covariate is not included in the MODEL statement.

You are interested in the same four fluid comparisons as in Example 50.1, shown in Table 50.10, except this time you want to marginalize over the effect of altitude.

For each of these contrasts, you want to determine the sample size required to achieve a power of 0.9 to detect an effect with magnitude according to Table 50.11. You are not yet attempting to choose a single sample size for the study, but rather checking the range of sample sizes needed by individual contrasts. You plan to test each contrast at $\alpha = 0.025$. You will provide twice as many runners with water as with any of the electrolytes, and you predict that you can study approximately two-thirds as many runners at high altitude than at low altitude. The resulting planned sample size weighting scheme is shown in Table 50.12. Since the scheme is only approximate, you use the NFRATIONAL option in the POWER statement to disable the rounding of sample sizes up to integers satisfying the weights exactly.

### Table 50.12 Approximate Sample Size Allocation Weights

<table>
<thead>
<tr>
<th>Fluid</th>
<th>Altitude</th>
<th>Water</th>
<th>EZD1</th>
<th>EZD2</th>
<th>LZ1</th>
<th>LZ2</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>High</td>
<td>4</td>
<td>2</td>
<td>2</td>
<td>2</td>
<td>2</td>
</tr>
<tr>
<td></td>
<td>Low</td>
<td>6</td>
<td>3</td>
<td>3</td>
<td>3</td>
<td>3</td>
</tr>
</tbody>
</table>

First, you create the exemplary data set to specify means and weights for the design profiles:

```sas
data Fluids2;
  input Altitude $ Fluid $ LacticAcid CellWgt;
datalines;
  High Water 36.9 4
  High EZD1 35.0 2
  High EZD2 31.5 2
  High LZ1 30 2
  High LZ2 27.1 2
  Low Water 34.3 6
  Low EZD1 32.4 3
```


Example 50.2: Two-Way ANOVA with Covariate

The variables Altitude, Fluid, and LacticAcid specify the factors and cell means in Table 50.11. The variable CellWgt contains the sample size allocation weights in Table 50.12.

Use the DATA= option in the PROC GLMPOWER statement to specify Fluids2 as the exemplary data set. The following statements perform the sample size analysis:

```sas
proc glmpower data=Fluids2;
  class Altitude Fluid;
  model LacticAcid = Altitude Fluid;
  weight CellWgt;
  contrast "Water vs. others" Fluid -1 -1 -1 -1 4;
  contrast "EZD vs. LZ" Fluid 1 1 -1 -1 0;
  contrast "EZD1 vs. EZD2" Fluid 1 -1 0 0 0;
  contrast "LZ1 vs. LZ2" Fluid 0 0 1 -1 0;
  power
    nfractional
    stddev = 3.5
    ncovariates = 1
    corrxy = 0.2 0.3 0
    alpha = 0.025
    ntotal = .
    power = 0.9;
run;
```

The CLASS statement identifies Altitude and Fluid as classification variables. The MODEL statement specifies the model, and the WEIGHT statement identifies CellWgt as the weight variable. The CONTRAST statement specifies the contrasts in Table 50.10. As in Example 50.1, the order of the contrast coefficients corresponds to the formatted class levels (EZD1, EZD2, LZ1, LZ2, Water). The POWER statement specifies total sample size as the result parameter and provides values for the other analysis parameters. The NCOVARIATES= option specifies the single covariate (body mass index), and the CORRXY= option specifies the two scenarios for its correlation with lactic acid buildup (0.2 and 0.3). Output 50.2.1 displays the results.

**Output 50.2.1** Sample Sizes for Two-Way ANOVA Contrasts

**The GLMPOWER Procedure**

<table>
<thead>
<tr>
<th>Fixed Scenario Elements</th>
<th>Value</th>
</tr>
</thead>
<tbody>
<tr>
<td>Dependent Variable</td>
<td>LacticAcid</td>
</tr>
<tr>
<td>Weight Variable</td>
<td>CellWgt</td>
</tr>
<tr>
<td>Alpha</td>
<td>0.025</td>
</tr>
<tr>
<td>Number of Covariates</td>
<td>1</td>
</tr>
<tr>
<td>Std Dev Without Covariate Adjustment</td>
<td>3.5</td>
</tr>
<tr>
<td>Nominal Power</td>
<td>0.9</td>
</tr>
</tbody>
</table>
Output 50.2.1  continued

| Index | Type  | Source     | Corr XY | Adj Std Dev | Test DF | Error DF | Fractional N Total | Actual Power | Ceiling N Total |
|-------|-------|------------|---------|-------------|---------|----------|--------------------|--------------|----------------|----------------|
| 1     | Effect| Altitude   | 0.2     | 3.43        | 1       | 84       | 90.418451          | 0.902        | 91             |
| 2     | Effect| Altitude   | 0.3     | 3.34        | 1       | 79       | 85.862649          | 0.901        | 86             |
| 3     | Effect| Altitude   | 0.0     | 3.50        | 1       | 88       | 94.063984          | 0.903        | 95             |
| 4     | Effect| Fluid      | 0.2     | 3.43        | 4       | 16       | 22.446173          | 0.912        | 23             |
| 5     | Effect| Fluid      | 0.3     | 3.34        | 4       | 15       | 21.687544          | 0.908        | 22             |
| 6     | Effect| Fluid      | 0.0     | 3.50        | 4       | 17       | 23.055716          | 0.919        | 24             |
| 7     | Contrast| Water vs. others | 0.2 | 3.43    | 1 | 15 | 21.720195 | 0.905 | 22 |
| 8     | Contrast| Water vs. others | 0.3 | 3.34    | 1 | 14 | 20.848805 | 0.903 | 21 |
| 9     | Contrast| Water vs. others | 0.0 | 3.50    | 1 | 16 | 22.422381 | 0.910 | 23 |
| 10    | Contrast| EZD vs. LZ | 0.2 | 3.43    | 1 | 35 | 41.657424 | 0.903 | 42 |
| 11    | Contrast| EZD vs. LZ | 0.3 | 3.34    | 1 | 33 | 39.674037 | 0.903 | 40 |
| 12    | Contrast| EZD vs. LZ | 0.0 | 3.50    | 1 | 37 | 43.246415 | 0.906 | 44 |
| 13    | Contrast| EZD1 vs. EZD2 | 0.2 | 3.43    | 1 | 139 | 145.613657 | 0.901 | 146 |
| 14    | Contrast| EZD1 vs. EZD2 | 0.3 | 3.34    | 1 | 132 | 138.173983 | 0.902 | 139 |
| 15    | Contrast| EZD1 vs. EZD2 | 0.0 | 3.50    | 1 | 145 | 151.565917 | 0.901 | 152 |
| 16    | Contrast| LZ1 vs. LZ2 | 0.2 | 3.43    | 1 | 268 | 274.055008 | 0.901 | 275 |
| 17    | Contrast| LZ1 vs. LZ2 | 0.3 | 3.34    | 1 | 253 | 259.919126 | 0.900 | 260 |
| 18    | Contrast| LZ1 vs. LZ2 | 0.0 | 3.50    | 1 | 279 | 285.363976 | 0.901 | 286 |

The sample sizes in Output 50.2.1 range from 21 for the comparison of water versus electrolytes (assuming a correlation of 0.3 between body mass and lactic acid buildup) to 275 for the comparison of LZ1 versus LZ2 (assuming a correlation of 0.2). PROC GLMPOWER also includes the effect tests for Altitude and Fluid. Note that the required sample sizes for this study are lower than those for the study in Example 50.1.
Note that the error standard deviation has been reduced from 3.5 to 3.43 (when correlation is 0.2) or 3.34 (when correlation is 0.3) in the approximation of the effect of the body mass index covariate. The error degrees of freedom has also been automatically adjusted, lowered by 1 (the number of covariates).

Suppose you want to plot the required sample size for the range of power values from 0.5 to 0.95. First, define the analysis by specifying the same statements as before, but add the PLOTONLY option to the PROC GLMPower statement to disable the nongraphical results. Next, specify the PLOT statement with X=POWER to request a plot with power on the X axis. Sample size is automatically placed on the Y axis. Use the MIN= and MAX= options in the PLOT statement to specify the power range. The following statements produce the plot:

```plaintext
ods graphics on;
proc glmpower data=Fluids2 plotonly;
  class Altitude Fluid;
  model LacticAcid = Altitude Fluid;
  weight CellWgt;
  contrast "Water vs. others" Fluid -1 -1 -1 -1 4;
  contrast "EZD vs. LZ" Fluid 1 1 -1 -1 0;
  contrast "EZD1 vs. EZD2" Fluid 1 -1 0 0 0;
  contrast "LZ1 vs. LZ2" Fluid 0 0 1 -1 0;
  power
    nfractional
    stddev = 3.5
    ncovariates = 1
    corrx = 0.2 0.3 0
    alpha = 0.025
    ntotal = .
    power = 0.9;
  plot x=power min=.5 max=.95;
run;
```

See Output 50.2.2 for the resulting plot.
In Output 50.1.2, the line style identifies the test, and the plotting symbol identifies the scenario for the correlation between covariate and response. The plotting symbol locations identify actual computed powers; the curves are linear interpolations of these points. As in Example 50.1, the required sample size is highest for the test of LZ1 versus LZ2.

Finally, suppose you want to plot the power for the range of sample sizes you will likely consider for the study (the range of 21 to 275 that achieves 0.9 power for different comparisons). In the POWER statement, identify power as the result (POWER=.), and specify NTOTAL=21. Specify the PLOT statement with X=N to request a plot with sample size on the X axis.

The following statements produce the plot:

```plaintext
proc glmpower data=Fluids2 plotonly;
  class Altitude Fluid;
  model LacticAcid = Altitude Fluid;
  weight CellWgt;
  contrast "Water vs. others" Fluid -1 -1 -1 -1 4;
  contrast "EZD vs. LZ" Fluid 1 1 -1 -1 0;
  contrast "EZD1 vs. EZD2" Fluid 1 -1 0 0 0;
  contrast "LZ1 vs. LZ2" Fluid 0 0 1 -1 0;
```
Example 50.2: Two-Way ANOVA with Covariate

```r
power
  nfractional
  stddev = 3.5
  ncovariates = 1
  corrxy = 0.2 0.3 0
  alpha = 0.025
  ntotal = 21
  power = .;
  plot x=n min=21 max=275;
run;

ods graphics off;
```

The `MAX=275` option in the `PLOT` statement sets the maximum sample size value. The `MIN=` option automatically defaults to the value of 21 from the `NTOTAL=` option in the `POWER` statement.

See **Output 50.2.3** for the plot.

**Output 50.2.3** Plot of Power versus Sample Size for Two-Way ANOVA Contrasts

Although **Output 50.2.2** and **Output 50.2.3** surface essentially the same computations for practical power ranges, they each provide a different quick visual assessment. **Output 50.2.2** reveals the range of required sample sizes for powers of interest, and **Output 50.2.3** reveals the range of powers achieved for sample sizes of interest.
Chapter 50: The GLMPOWER Procedure

Example 50.3: Repeated Measures ANOVA

Logan, Baron, and Kohout (1995) and Guo et al. (2013) study the effect of a dental intervention on the memory of pain after root canal therapy. The intervention is a sensory focus strategy, in which patients are instructed to pay attention only to the physical sensations in their mouth during the root canal procedure.

Suppose you are interested in the long-term effects of this sensory focus intervention, because avoidance behavior has been shown to build along with memory of pain. You are planning a study to compare sensory focus to standard of care over a period of a year, asking patients to self-report their memory of pain immediately after the procedure and then again at 1 week, 6 months, and 12 months. You use a scale from 0 (no pain remembered) to 5 (maximum pain remembered).

The between-subject factor in your model is treatment, with two levels (sensory focus versus standard of care), and you allocate each treatment equally for a balanced design. The within-subject factor is time, with four levels (0, 1, 26, and 52 weeks).

You want to determine the number of patients who are needed in order to achieve a power of 0.9 at significance level $\alpha = 0.01$ for the test of the interaction between time and treatment, where the contrast over time contains all pairwise comparisons. You also want to generate a plot of power versus sample size that covers the power range of 0.05 to 0.99.

The default Hotelling-Lawley $F$ test is appropriate for this study, especially because it is the same as the Wald test in PROC MIXED with the DDFM=KR Kenward-Roger degrees-of-freedom method and an unstructured covariance model.

You conjecture that the mean memory of pain for each treatment follows the information in Table 50.13.

Table 50.13  Mean Memory of Pain by Treatment

<table>
<thead>
<tr>
<th>Treatment</th>
<th>Time Since Root Canal Therapy</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>Later on Same Day</td>
</tr>
<tr>
<td>Sensory Focus</td>
<td>2.40</td>
</tr>
<tr>
<td>Standard of Care</td>
<td>2.40</td>
</tr>
</tbody>
</table>

The following statements create a data set named Pain that is to contain these means over treatment and time:

```plaintext
data Pain;
  input Treatment $ PainMem0 PainMem1Wk PainMem6Mo PainMem12Mo;
datalines;
  SensoryFocus  2.40  2.38  2.05  1.90
  StandardOfCare 2.40  2.39  2.36  2.30
;```

The variable Treatment specifies the two treatments. The four variables PainMem0, PainMem1Wk, PainMem6Mo, and PainMem12Mo specify the mean memory of pain scores in Table 50.13.

To characterize the variability, you must specify a set of parameters that defines the entire covariance matrix of the residuals. You conjecture that the error standard deviation is the same at all four time points, with a value somewhere between 0.92 and 1.04, and you account for your uncertainty by including both the lower and upper ends of this range in the sample size analysis. You believe that the correlation has a linear exponent autoregressive (LEAR) structure, with a correlation of about 0.6 between measurements one week apart and a
decay rate of about 0.8 over one-week intervals. The correlation matrix that contains these LEAR parameters, rounded to three decimal places, is shown in Table 50.14.

<table>
<thead>
<tr>
<th></th>
<th>0</th>
<th>1</th>
<th>26</th>
<th>52</th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>1</td>
<td>0.6</td>
<td>0.491</td>
<td>0.399</td>
</tr>
<tr>
<td>1</td>
<td>0.6</td>
<td>1</td>
<td>0.495</td>
<td>0.402</td>
</tr>
<tr>
<td>26</td>
<td>0.491</td>
<td>0.495</td>
<td>1</td>
<td>0.491</td>
</tr>
<tr>
<td>52</td>
<td>0.399</td>
<td>0.402</td>
<td>0.491</td>
<td>1</td>
</tr>
</tbody>
</table>

Use the DATA= option in the PROC GLMPOWER statement to specify Pain as the exemplary data set. Specify the between- and within-subject factors and the model by using the CLASS, MODEL, and REPEATED statements just as you would in PROC GLM for the repeated measures data analysis. Use the POWER statement to indicate sample size as the result parameter and specify the other analysis parameters, and use the PLOT statement to generate the power curves. The following statements perform the sample size analysis:

```plaintext
ods graphics on;
proc glmpower data=Pain;
  class Treatment;
  model PainMem0 PainMem1Wk PainMem6Mo PainMem12Mo = Treatment;
  repeated Time contrast;
  power
    mtest = hlt
    alpha = 0.01
    power = .9
    ntotal = .
    stddev = 0.92 1.04
    matrix ("PainCorr") = lear(0.6, 0.8, 4, 0 1 26 52)
    corrmat = "PainCorr";
  plot y=power min=0.05 max=0.99 yopts=(ref=0.9)
    vary (linestyle by stddev, symbol by dependent source);
run;
ods graphics off;
```

The STDDEV= option specifies the two scenarios for the common residual standard deviation, 0.92 and 1.04. The MATRIX= option defines the LEAR correlation structure, and the CORRMAT= option specifies it as the correlation matrix of the residuals. The Y=POWER option in the PLOT statement requests a plot that has power on the Y axis. (The result parameter—in this case, total sample size—is always plotted on the other axis.) The MIN= and MAX= options in the PLOT statement specify the power range. The YOPTS=(REF=) option adds a reference line at the target power value of 0.9. The VARY option specifies that the line style vary by the residual standard deviation and that the plotting symbol vary by the combination of within-subject and between-subject effects. The ODS GRAPHICS ON statement enables ODS Graphics.

Output 50.3.1 shows the output, and Output 50.3.2 shows the plot.
Output 50.3.1 reveals that the required sample size to achieve a power of 0.9 for the test of the Time*Treatment interaction is 350 for the error standard deviation of 0.92 and 446 for the error standard deviation of 1.04.
References


actual power
  GLMPOWER procedure, 3947, 3949, 3963
alpha level
  GLMPOWER procedure, 3937
analysis of covariance
  power and sample size (GLMPOWER), 3967
analysis of variance
  multivariate (GLMPOWER), 3929, 3943
  power and sample size (GLMPOWER), 3919, 3961, 3967
between-subject factors
  repeated measures, 3918, 3925, 3928, 3953
ceiling sample size
  GLMPOWER procedure, 3949
classification variables
  GLMPOWER procedure, 3928, 3931
contrasts
  power and sample size (GLMPOWER), 3923, 3928, 3951, 3953, 3961, 3967
  repeated measures (GLMPOWER), 3945
correlation
  GLMPOWER procedure, 3937, 3938, 3943
covariance
  GLMPOWER procedure, 3938, 3943
covariates
  GLMPOWER procedure, 3931, 3938, 3941–3943, 3952, 3967
effect
  GLMPOWER procedure, 3949
exemplary data set
  power and sample size (GLMPOWER), 3918, 3920, 3927, 3931, 3946, 3947, 3962
fractional sample size
  GLMPOWER procedure, 3949
GLMPOWER procedure
  actual alpha, 3949
  actual power, 3947, 3949, 3963
  alpha level, 3937
  analysis of variance, 3919, 3961, 3967
  ceiling sample size, 3949
  compared to other procedures, 3919
  computational methods, 3950
  contrasts, 3923, 3928, 3945, 3951, 3953, 3961, 3967
correlation, 3937, 3938, 3943
covariance, 3938, 3943
covariates, class and continuous, 3931, 3938, 3941–3943, 3952, 3967
displayed output, 3949
effect, 3949
exemplary data set, 3918, 3920, 3927, 3931, 3946, 3947, 3962
fractional sample size, 3949
graphics, 3961
introductory example, 3919
keyword-lists, 3946
Kronecker product, 3939
linear exponent autoregressive (LEAR), 3939
multivariate analysis of variance, 3929, 3943, 3974
name-lists, 3946
nominal power, 3947, 3949, 3963
number-lists, 3946
ODS graph names, 3961
ODS Graphics, 3961
ODS table names, 3950
ordering of effects, 3927
plots, 3919, 3925, 3927, 3931
positional requirements for statements, 3925
repeated measures, 3929, 3943, 3974
repeated measures examples, 3946
sample size adjustment, 3947
standard deviation, 3938, 3942, 3943
statistical graphics, 3961
summary of statements, 3926
transformations for MANOVA, 3930
transformations for repeated measures, 3945
value lists, 3946
variance, 3938, 3942
graphics
  GLMPOWER procedure, 3961
keyword-lists
  GLMPOWER procedure, 3946
Kronecker product
  GLMPOWER procedure, 3939
LEAR, see linear exponent autoregressive (LEAR)
linear exponent autoregressive (LEAR)
  GLMPOWER procedure, 3939
MANOVA, see multivariate analysis of variance
multivariate analysis of variance
GLMPOWER procedure, 3929, 3943
power and sample size (GLMPOWER), 3974

name-lists
  GLMPOWER procedure, 3946
nominal power
  GLMPOWER procedure, 3947, 3949, 3963
number-lists
  GLMPOWER procedure, 3946

ODS graph names
  GLMPOWER procedure, 3961
ODS Graphics
  GLMPOWER procedure, 3961
orthogonalizing transformation matrix
  GLMPOWER procedure, 3931
plots
  power and sample size (GLMPOWER), 3919,
    3925, 3927, 3931
power
  See GLMPOWER procedure, 3917
POWER procedure
  compared to other procedures, 3919
prospective power, 3918

repeated measures
  contrasts (GLMPOWER), 3945
  examples (GLMPOWER), 3946
  GLMPOWER procedure, 3929, 3943
  power and sample size (GLMPOWER), 3974
retrospective power, 3918

sample size
  See GLMPOWER procedure, 3917
sample size adjustment
  GLMPOWER procedure, 3947
standard deviation
  GLMPOWER procedure, 3938, 3942, 3943
statistical graphics
  GLMPOWER procedure, 3961

transformation matrix
  orthogonalizing, 3931
transformations
  for multivariate ANOVA, 3930
transformations for repeated measures
  GLMPOWER procedure, 3945

value lists
  GLMPOWER procedure, 3946
variance
  GLMPOWER procedure, 3938, 3942

within-subject factors
  repeated measures, 3919, 3926, 3944, 3953
Syntax Index

ALPHA= option
  POWER statement (GLMPOWER), 3937

BY statement
  GLMPOWER procedure, 3927

CLASS statement
  GLMPOWER procedure, 3928

CONTRAST option
  REPEATED statement (GLMPOWER), 3945

CORRMAT option
  POWER statement (GLMPOWER), 3937

CORRS option
  POWER statement (GLMPOWER), 3937

CORRXY= option
  POWER statement (GLMPOWER), 3938

COVMAT option
  POWER statement (GLMPOWER), 3938

DATA= option
  PROC GLMPOWER statement, 3927

DEPENDENT option
  POWER statement (GLMPOWER), 3938

DESCRIPTION= option
  PLOT statement (GLMPOWER), 3935

EFFECTS option
  POWER statement (GLMPOWER), 3938

factor specification
  REPEATED statement (GLMPOWER), 3944

GLMPOWER procedure
  syntax, 3925

GLMPOWER procedure, BY statement, 3927
GLMPOWER procedure, CLASS statement, 3928
GLMPOWER procedure, CONTRAST statement, 3928
  SINGULAR= option, 3929
GLMPOWER procedure, MANOVA statement, 3929
  M= option, 3930
  ORTH option, 3931
GLMPOWER procedure, MODEL statement, 3931
GLMPOWER procedure, PLOT statement, 3931
  DESCRIPTION= option, 3935
  INTERPOL= option, 3935
  KEY= option, 3933

MARKERS= option, 3933
MAX= option, 3934
MIN= option, 3934
NAME= option, 3935
NPOINTS= option, 3934
STEP= option, 3934
VARY option, 3934
X= option, 3934
XOPTS= option, 3935
Y= option, 3935
YOPTS= option, 3935

GLMPOWER procedure, POWER statement, 3936
  ALPHA= option, 3937
  CORRMAT option, 3937
  CORRS option, 3937
  CORRXY= option, 3938
  COVMAT option, 3938
  DEPENDENT option, 3938
  EFFECTS option, 3938
  MATRIX option, 3938
  METHOD= option, 3940
  MTEST= option, 3940
  NCOVARIATES= option, 3941
  NFRACTIONAL option, 3941
  NTOTAL= option, 3941
  OUTPUTORDER= option, 3942
  POWER= option, 3942
  PROPVARREDUCTION= option, 3942
  SQRTVAR option, 3942
  STDDEV= option, 3943
  UEPSDEF= option, 3943

GLMPOWER procedure, PROC GLMPOWER statement, 3926
  DATA= option, 3927
  ORDER= option, 3927
  PLOTONLY= option, 3927

GLMPOWER procedure, REPEATED statement, 3943
  CONTRAST option, 3945
  factor specification, 3944
  HELMERT option, 3945
  IDENTITY option, 3945
  MEAN option, 3945
  POLYNOMIAL option, 3946
  PROFILE option, 3946

GLMPOWER procedure, WEIGHT statement, 3946

HELMERT option
  REPEATED statement (GLMPOWER), 3945
IDENTITY option
     REPEATED statement (GLMPOWER), 3945
INTERPOL= option
     PLOT statement (GLMPOWER), 3933
KEY= option
     PLOT statement (GLMPOWER), 3933
M= option
     MANOVA statement (GLMPOWER), 3930
     MANOVA statement
          GLMPOWER procedure, 3929
MARKERS= option
     PLOT statement (GLMPOWER), 3933
MATRIX option
     POWER statement (GLMPOWER), 3938
MAX= option
     PLOT statement (GLMPOWER), 3934
MEAN option
     REPEATED statement (GLMPOWER), 3945
METHOD= option
     POWER statement (GLMPOWER), 3940
MIN= option
     PLOT statement (GLMPOWER), 3934
MODEL statement
     GLMPOWER procedure, 3931
MTEST= option
     POWER statement (GLMPOWER), 3940
NAME= option
     PLOT statement (GLMPOWER), 3935
NCOVARIATES= option
     POWER statement (GLMPOWER), 3941
NFRACTIONAL option
     POWER statement (GLMPOWER), 3941
NPOINTS= option
     PLOT statement (GLMPOWER), 3934
NTOTAL= option
     POWER statement (GLMPOWER), 3941
ORDER= option
     PROC GLMPOWER statement, 3927
ORTH option
     MANOVA statement (GLMPOWER), 3931
OUTPUTORDER= option
     POWER statement (GLMPOWER), 3942
POWER= option
     POWER statement (GLMPOWER), 3942
PROC GLMPOWER statement, see GLMPOWER procedure
PROFILE option
     REPEATED statement (GLMPOWER), 3946
PROPVARREDUCTION= option
     POWER statement (GLMPOWER), 3942
REPEATED statement
     GLMPOWER procedure, 3943
SINGULAR= option
     CONTRAST statement (GLMPOWER), 3929
SQRTVAR option
     POWER statement (GLMPOWER), 3942
STDDEV= option
     POWER statement (GLMPOWER), 3943
STEP= option
     PLOT statement (GLMPOWER), 3934
UEPSDEF= option
     POWER statement (GLMPOWER), 3943
VARY option
     PLOT statement (GLMPOWER), 3934
WEIGHT statement
     GLMPOWER procedure, 3946
X= option
     PLOT statement (GLMPOWER), 3934
XOPTS= option
     PLOT statement (GLMPOWER), 3935
Y= option
     PLOT statement (GLMPOWER), 3935
YOPTS= option
     PLOT statement (GLMPOWER), 3935

PLOT statement
     GLMPOWER procedure, 3931
PLOTONLY= option
     PROC GLMPOWER statement, 3927
POLYNOMIAL option
     REPEATED statement (GLMPOWER), 3946
POWER statement
     GLMPOWER procedure, 3936